
                          HEATH DISK OPERATING SYSTEM 
 
                           SOFTWARE REFERENCE MANUAL 
 
                                 VERSION 3.02 
 
 
    HDOS  was originally copyrighted in 1980 by the Heath Company.  Through 
    the years it continued to be improved  by  successive  revisions  which 
    included  1.5, 1.6, and finally 2.0.  It was entered into public domain 
    on 19 July 1989 per letter by Jim Buszkiewicz, Managing  Editor,  Heath 
    Users'   Group,   P.O.    Box   217,   Benton   Harbor,  MI  49022-0217 
    (616)982-3463.   A  copy  of  this  letter  is  available  for   public 
    inspection.  Indeed, HDOS is still alive and well! 
 
    This  manual is indicative of further improvements and provides for the 
    latest revision, HDOS 3.0 and HDOS 3.02.  Revision 3.0 is  detailed  in 
    chapters  1,  2,  and  3,  while chapters 4, 5, 6, 7, 8, 13 and 14, are 
    related  to  revision  3.02.   Chapters  9  through  12,   with   minor 
    improvements,  are  essentially  picked  up  from the original HDOS 2.0 
    manual.  
 
    Chapter 13, The HDOS Programmers' Reference Manual, is intended for the 
    advanced programmer.  This chapter calls out and describes all  of  the 
    new  .SCALLS  for HDOS 3.02.  You will find some references to HDOS 2.0 
    and its overlays in this chapter.  These should be ignored. 
     
    SPECIAL  DISCLAIMER:  The  Heath Company cannot provide consultation on 
    either the HDOS Operating System or user-developed or modified versions 
    of Heath software products designed to operate under the HDOS Operating 
    System.  Do not refer to Heath for questions. 
 
    Instead,  you  are invited to direct any questions concerning the Heath 
    Disk Operating System (HDOS) to Mr.  Kirk L.  Thompson, Editor "Staunch 
    89/8"  Newsletter,  P.O.   Box 548, #6 West Branch Mobile Home Village, 
    West Branch, IA 52358. 

                             PART 1 - INTRODUCTION 
                             +++++++++++++++++++++ 
 
    Purpose 
    ======= 
 
    This  manual  describes  the  advanced features of HDOS version 3.0 and 
    3.02 that are necessary for a user program to interface  with  HDOS  at 
    the  assembly  language  level.   This  information is provided for the 
    advanced programmer.  References to overlays pertain to HDOS 2.0. 
     
    Data and information provided in this manual applies equally to the H8, 
    H89, or Z90 family of computers. 
 
 
    Background 
    ========== 
 
    Chapter 11, ASM, of this "HDOS Software Reference Manual" documents the 
    various system commands and  BASIC  statements  used  to  generate  and 
    maintain files at the higher language level.  At this level, the novice 
    or average programmer need not be concerned about the involved  details 
    of interfacing his programs with HDOS or the disk drives. 
     
 
    Preface 
    ======= 
 
    HDOS provides a full run-time support environment for assembly language 
    programs.    Communications   with   file-oriented   devices,   console 
    communications, memory allocation, and other such services are provided 
    by the HDOS  system.   Since  the  H8  does  not  afford  any  hardware 
    protection,  assembly  language routines must be "polite," in that they 
    should not damage the H8 running environment.   This  subject  will  be 
    discussed in greater detail further on in this document.           
 
 
    File Names 
    ========== 
 
    Since  many SCALLs require file names as arguments, this is a good time 
    to discuss HDOS file names. 
     
    In general, when you supply a file name as an argument to a SCALL,  you 
    point  to  an  ASCII  string containing the file descriptor just as the 
    user would have typed  it.   The  line  should  be  terminated  with  a 
    delimiter  of  some  sort,  usually  a  comma,  blank, or zerozero (00) 
    character.  For example, the  following  are  examples  of  valid  file 
    names: 
 
             DB      'SY0:MYFILE.TMP',0 
             DB      'TMP',0 
             DB      'BASIC.SAM,'   

                         PART 1 - INTRODUCTION (Cont) 
                         ++++++++++++++++++++++++++++ 
 
    File Names (Cont) 
    ================= 
 
    NOTE: The expression ',' delimits the file name. 
 
    Of  course  these names are all shown being assembled into the program. 
    You might just as well have read  them  from  the  user's  console,  or 
    generated  the  names  on the keyboard.  They must not have imbedded 00 
    bytes or blanks in the names. 
     
    Also note that some of the examples shown do not specify  an  extension 
    or a device.  All SCALLs that take file names as arguments also require 
    a default block.  This block is a 6-byte area  containing  the  default 
    device  specification and a default extension specification.  A typical 
    default block is: 
 
             DB      'SY0 TMP' 
 
    which  yields  a  default device of SY0: and extension of TMP.  Another 
    common block is: 
 
             DB      'SY0',0,0,0 
 
    which  indicates  that there is no default extension.  File descriptors 
    not specifying a name will generate a file with a null extension. 
    *********************************************************************** 
 
 
                         PART 2 - RUN-TIME ENVIRONMENT 
                         +++++++++++++++++++++++++++++ 
 
    Memory Layout  
    ============= 
 
    HDOS  contains  many  useful  general-purpose  subroutines which may be 
    called by user programs.  These,  together  with  the  system  services 
    provided,   make   assembly   language   programming  under  HDOS  very 
    convenient. 
 
    When  you type "RUN fname," HDOS will load your program into memory and 
    run it.  This section will discuss the initial run-time environment  of 
    the  program.   Refer  to the memory map in the HDOS Software Reference 
    Manual, Chapter 8, Appendix 8-A, and Appendix  13-D, for  further  data 
    that  will  help  you  to  understand  the information provided in this 
    chapter. 
     
    The first 64 bytes of RAM from 040000 to 040100 are used by PAM-8.  The 
    PAM-8 source listing documents their use.            

                     PART 2 - RUN-TIME ENVIRONMENT (Cont) 
                     ++++++++++++++++++++++++++++++++++++ 
 
    Memory Layout (Cont) 
    ==================== 
 
                                     NOTE 
              Although   the   PAM-8   ROM   will  be  referenced 
              throughout   this   manual,   the   general-purpose 
              routines  of the MTR -88, MTR-89, MTR-90, PAM-8-GO, 
              and XCON-8 ROMs all have common entry points.   For 
              specific  information,  refer  to  the  data in the 
              manual covering the ROM that you are using.        
               
    The next 295 bytes are used by HDOS and the disk device driver for work 
    cells.  These cells are in low memory, so that HDOS can reference  them 
    without  having  to  compute relocation factors (HDOS is relocatable in 
    low memory).  Some of the contents of these cells are  of  interest  to 
    assembly  language  programmers  and  are  indirectly available through 
    HDOS system calls.  You should refrain from  accessing  them  directly, 
    since  their  position  may  change  with  future releases.  Use of the 
    proper HDOS symbols and system calls in assembly language programs will 
    make  it  possible  to  transport  your  programs  to future Heath CPUs 
    executing an advanced HDOS system.  There are a few cells that  may  be 
    of  interest  to  the  programmer,  and these are documented in Part 6. 
    They may be read, but never written to. 
     
    Following  the  work  cell  area is a 279-byte stack area.  When a user 
    program is executed, the stack pointer is  set  to  the  symbol  STACK, 
    which  is  042200A.  Note that you may not set your stack pointer below 
    that  address and then use the area  below  42200A  for  code  or  data 
    (other  than  that  stored  by  a normal PUSH).  You may make the stack 
    larger, setting SP to a value larger than 042200A.  Calls to  the  HDOS 
    system will preserve this larger stack.              
     
    The  user  program area starts at 042200A, immediately after the top of 
    the stack.  The user program extends until the last byte loaded by  the 
    RUN  command.  Note that the assembler generates a dummy 00 byte as the 
    last statement in a program, so that trailing DS declarations  will  be 
    contained  in  the size of the running program.  There is a system call 
    which requests access to more memory.  You must issue the  call  first, 
    since HDOS may be using that area for its own code.              
 
    Any  active  device  drivers are loaded immediately before the resident 
    HDOS code.  A device driver is loaded when a file is opened on a device 
    whose  driver  is  not  yet  in  memory.   The  TT:  device  driver  is 
    automatically loaded during the boot process, and  never  needs  to  be 
    loaded.   Since  the  SY: driver is permanently loaded into memory when 
    the system is first booted up, it never needs to be loaded. 

                     PART 2 - RUN-TIME ENVIRONMENT (Cont) 
                     ++++++++++++++++++++++++++++++++++++ 
 
    Memory Layout (Cont) 
    ==================== 
      
    Finally,  the  HDOS  system  resides in low memory.  When the system is 
    booted up, HDOS initially loads at a fixed lower address.  After sizing 
    memory,  HDOS  moves  its  permanently  resident parts into low memory. 
    This section contains the SCALL Dispatcher and  the  handlers  for  all 
    standard SCALL functions. 
 
    I/O Environment 
    =============== 
 
                                  TABLE 13-1 
                      PORT ASSIGNMENTS FOR THE H89 AND H8 
    +---------------------+-----------------------------------------------+ 
    |       PORT          |   H89 COMPUTER **** BOTH ***** H8 COMPUTER    | 
    +---------------------+-----------------------------------------------+ 
    | 170-173Q (078-078H) | ---- H37, H47, or H67 Disk Drive Systems ---- | 
    | 174-177Q (07C-07FH) | ---- H17, H47, or H67 Disk Drive Systems ---- | 
    | 300-307Q (0C0-0C7H) | ---------------- Reserved ------------------- | 
    | 320-327Q (0D0-0D7H) | H88-3 ------- Alternate Terminal ------- H8-4 | 
    | 330-337Q (0D8-0DFH) | H88-3 ------------  Modem -------------- H8-4 | 
    | 340-347Q (0E0-0E7H) | H88-3 ---------- Line Printer ---------- H8-4 | 
    | 350-357Q (0E8-0EFH) | CPU ---------- Console Terminal -------- H8-4 | 
    | 360-361Q (0F0-0F1H) | ---- Reserved ----     --- H8 Front Panel --- | 
    | 370-371Q (0F8-0F9H) | H88-5 ------------ Cassette ------------ H8-5 | 
    | 372-373Q (0F2-0FBH) | ---- Reserved ----     ----- Console Terminal |   
    | 374-375Q (0FC-0FDH) | ---- Reserved ----     --- Alternate Terminal | 
    | 376-377Q (0FE-0FFH) | ---- Reserved ----     ------  Reserved ----- | 
    +---------------------+-----------------------------------------------+ 
                                                                            
    HDOS  has  a  vested interest in the I/O ports being used by the device 
    drivers currently in memory.  These ports should not be disturbed  when 
    HDOS  or  a  device  driver  may  be trying to use them.  The ports are 
    listed in Table 13-1.         
 
    Since  the  TT:  and  SY: drivers are permanently resident, it is vital 
    that you do not disturb the TT: and SY: ports.  Disturbing the SY: port 
    will destroy your system disk.  Disturbing the TT: port will damage the 
    console driver package.  The console driver package  communicates  with 
    the console device at interrupt time, so you will not be able to detect 
    character entry by examining the console status  bits.   HDOS  provides 
    you with a facility to test the presence of a console character. 
     
 
    Interrupt Environment 
    ===================== 
 
    HDOS  is  an  interrupt-driven  system,  so  be  careful how you handle 
    interrupts.  Your program must not  turn  off  interrupts  via  the  DI 

                     PART 2 - RUN-TIME ENVIRONMENT (Cont) 
                     ++++++++++++++++++++++++++++++++++++ 
 
    Interrupt Environment (Cont) 
    ============================ 
 
    instruction  for other than very short periods of time.  The H17 device 
    driver makes use  of  the  front  panel  clock  interrupts  on  the  H8 
    computer,  so  you must not disable them, either directly via port 360Q 
    or by the PAM-8 control word.  Likewise, console interrupts are used by 
    the system console handler, and should not be disturbed.  HDOS does not 
    currently support any interrupt-driven device drivers  except  the  H37 
    device  driver,  but  programs may still make use of interrupts.  There 
    are  two  major  trouble  areas  in  this:  choosing  a   vector    and 
    discontinuing the interrupts. 
     
 
    Interrupt Vectors 
    ----------------- 
 
    Of  the  eight  interrupt  vectors available in an 8080A microprocessor 
    (sometimes abbreviated "processor") HDOS makes use of six or  seven  of 
    them.                                                                 
     
         0 ----- Master Clear.  Returns control to PAM-8. 
         1 ----- Clock Interrupts. 
         2 ----- Single-Step.  Used by DBUG.  May be used by a user 
                   program when not running DBUG. 
         3 ----- Console Interrupts. 
         4 ----- Reserved for H37. 
         4 ----- Reserved for H14 Printer. 
         5 ----- Reserved for modems. 
         5 ----- Available for user programs. 
         6 ----- Available for user programs. 
         7 ----- HDOS SCALL vector.                     
 
    Set  the  vectors by storing a JMP to your interrupt service routine in 
    the PAM-8 ".UIVEC" area, as discussed in the PAM-8 manual.  
 
 
    Discontinuing Interrupts 
    ------------------------ 
 
    When  a  user  program  causes a device to start issuing interrupts, it 
    must somehow turn off that device before control returns to the system. 
    HDOS  will not alter the interrupt vector (JMP) in PAM-8's "UIVEC," and 
    an interrupt occurring after your program has  been  removed  from  the 
    system  will  be tragic.  Also note that as a user, you must be careful 
    of typing CTRL-Z, as this can kill your program before it can shut down 
    any interrupting devices.  

                     PART 2 - RUN-TIME ENVIRONMENT (Cont) 
                     ++++++++++++++++++++++++++++++++++++ 
 
    Discontinuing Interrupts (Cont) 
    ------------------------------- 
     
                                     NOTE 
              You  MUST  turn  off  the  device interrupts before 
              surrendering control  to  HDOS.   Simply  replacing 
              your  interrupt vector with EI and RET instructions 
              will cause disaster, since the interrupting  device 
              will  continue  to  request  interrupts until it is 
              serviced, and HDOS does not know how to service it. 
              Your  computer  will  then  hang  in  an  interrupt 
              service loop.  
               
 
    CPU Environment 
    =============== 
 
    After  loading  your  program,  HDOS transfers control to the program's 
    entry point.  This is the address  specified  in  the  END  (assembler) 
    pseudo.  
     
 
    Channel Environment 
    =================== 
 
    HDOS allows user programs to communicate with file-oriented devices via 
    "channels."  These channels are discussed in Part  3.   In  all  cases, 
    channel  -1  (377Q) is open for read access on the device and file that 
    the program was loaded from.  This is done so you can conveniently load 
    overlays  without  having  to  know under what name and disk drive your 
    program was run from.  If your program was run  in  response to  a  RUN 
    command, all other channels will be closed.  If your program was run in 
    response  to a .LINK SCALL, then the other channels will remain as they 
    were set up by the program which issued the .LINK.  
    *********************************************************************** 
 
 
                             PART 3 - I/O CHANNELS 
                             +++++++++++++++++++++ 
 
    All  file  I/O in the HDOS system is done via I/O channels.  "File I/O" 
    refers to normal input/output done to  HDOS  devices  via  HDOS  device 
    drivers.  Naturally, a program may control its "private devices" (those 
    not suitable for device drivers) in any way it pleases. 
     
    In general, the sequence for doing file I/O is to issue an "open" SCALL 
    (i.e., .OPENR, .OPENW, or .OPENU) to HDOS, supplying HDOS with the file 
    descriptor  as  an  ASCII string.  HDOS will parse the string, load the 
    device driver (if necessary), and open the file.  When  you  issue  the 
    "OPEN"  SCALL,  you  supply a channel number from -1 (i.e., 377Q) to 5. 

                         PART 3 - I/O CHANNELS (Cont) 
                         ++++++++++++++++++++++++++++ 
 
    This channel number must not already be in use.  This  means  that  you 
    may open a maximum of seven files simultaneously.  
     
    Once  the file has been opened, you can perform I/O by using the .READ, 
    .WRITE, and .POSIT SCALLs.  Make these requests by supplying HDOS  with 
    the  channel  number  of  the file you want read or written.  After the 
    initial open, you no longer need the file  descriptor  string.   Should 
    you  suddenly  need  that  file  name (for  example, to  issue an error 
    message), HDOS provides the .NAME SCALL to recall the  file  name  used 
    when that channel was opened.  
 
    All  disk  file  I/O is done in multiples of 256 bytes, the HDOS system 
    sector size.  As many bytes as desired may be transferred at one  time, 
    so  long  as  the  count  is an integer multiple of 256.  HDOS normally 
    performs I/O in a sequential fashion.  For example, if your program  is 
    reading  from  a  disk file one sector (256 bytes) at a time, the first 
    read will return sector 0, the next read sector 1, etc.  For each  open 
    channel, HDOS maintains a "sector cursor," which indicates which sector 
    in the file is next to be read  or  written.   HDOS  does  provide  the 
    facility,  via  .POSIT, to randomly read and write sectors to or from a 
    disk file by changing the value of this "sector cursor."  
     
    When  you  are  done  with  the  file,  use the .CLOSE SCALL, once more 
    supplying the channel number.  HDOS will close the file and  thus  make 
    that channel available for another open command.  
     
                                     NOTE 
              Although  channel  -1  may  be  used  as  a general 
              purpose I/O channel, its  use  should  normally  be 
              avoided.   It is already open when your  program is 
              started, but you must close it before you can  open 
              a  file  on  it.   Also, channel -1 will be cleared 
              (see the .CLEAR SCALL for details) if you  use  the 
              .LINK  SCALL.   Thus,  any  file  open for write on 
              channel -1 at that time will be lost.            
    ********************************************************************** 
 
                             PART 4 - PRECAUTIONS 
                             ++++++++++++++++++++ 
 
    We  have  stated earlier in this document that the HDOS system does not 
    provide any hardware protection, and thus is vulnerable  to  errors  in 
    assembly  language  programs.   In order to help minimize this problem, 
    this segment discusses the  "Do's  and  Dont's"  of  assembly  language 
    programming in greater detail.             

    Memory Precautions 
    ================== 
 
    The two most important areas of memory precautions are: respect for the 
    user program area and maintenance of the stack.            
 
 
    User Memory Area 
 
    ---------------- 
 
    A  user  program  must  never  write into memory outside of its domain. 
    This "domain" consists of the memory area from 042200A (USERFWA) to the 
    LWA  of the user program area.  When your program is first loaded, this 
    LWA is set up to the end of your program and its  declared  data  areas 
    via DS, DW, or DB; not EQU.  The ".SETTOP" SCALL is available to adjust 
    this limit.  User programs may adjust this limit as often as they  like 
    (for  details,  refer  to  the .SETTOP SCALL documentation).  Note that 
    HDOS may use all memory after this limit for a storage area,  which  is 
    going to cause trouble if your routine also tries to use it.            
 
 
    Stack Maintenance 
    ----------------- 
 
    Since  the  HDOS  system  uses  interrupts and  requires  interrupts to 
    handle the console, the H17 disk, and the H37 disk, your program may be 
    interrupted  at  any  time.   You  must  always  maintain a valid stack 
    pointer  with at least 64 free bytes on the top of that stack.  If  you 
    plan  to  fill  the system stack area, then you should ORG your program 
    above 042200A and set the stack pointer  higher,  giving  yourself  and 
    HDOS  a  bigger stack.  HDOS does not use a separate stack; it uses the 
    top of the user program stack. 
     
 
    I/O PRECAUTIONS 
    =============== 
 
    As  we  discussed  earlier, I/O precautions consist of keeping your INs 
    and OUTs to yourself.  Don't disturb the H17, the H37, and the H47, and 
    don't disturb the console ports!  Also, be careful what you do with the 
    front panel ports on the H8, either directly or indirectly  via  PAM-8. 
    These  ports  control the clock interrupts, which are necessary for the 
    H17 device driver.              
 
     
    INTERRUPT PRECAUTIONS 
    ===================== 
 
    When you are using interrupts, you must use only the available vectors, 
    which are 4 (if you are not using the H37), 5 (if you are not 

                         +++++++++++++++++++++++++++ 
 
    INTERRUPT PRECAUTIONS (Cont) 
    ============================ 
 
    using  a  modem),  and 6.  You may also use 2, if you will not be using 
    DBUG.  Before you enable your interrupting device, install the  service 
    vector in the appropriate ".UIVEC" location. 
     
    Most  importantly,  turn off the interrupting device so it cannot issue 
    any  more interrupts before you either return control to HDOS or  CTL-Z 
    out  of  the  program.   If an interrupt occurs when your program is no 
    longer there to service it,  the  operating  system  and  possibly  the 
    information on your diskettes will be destroyed.             
     
    Since  console and clock interrupts may occur at any time, your program 
    should not turn off interrupts via DI, except for very short periods of 
    time.            
     
    Finally,  HDOS uses the clock interrupts, so you should not overlay its 
    interrupt vector.  Programs desiring clock service should use all means 
    possible  to  make  do with the interrupt counter (PAM-8s .TICCNT).  If 
    you absolutely must have clock interrupts,  save  the  address  in  the 
    clock  vector,  install  your own vector, and have your service routine 
    exit the interrupt by jumping to the HDOS vector  address.   HDOS  uses 
    the  clock  interrupts  for H17 timings; disturbing it might cause your 
    motors to keep spinning, prematurely wearing the motors.  Or worse, you 
    might defeat the H17 driver's head-settle delays and cause a bad sector 
    to be written. 
     
 
    CPU PRECAUTIONS 
    =============== 
 
    This   precaution   should   be   familiar  to  all  assembly  language 
    programmers: don't let the  CPU  execute  undefined  memory  locations. 
    Should  such  a  thing  occur,  it  is unlikely that your disks will be 
    damaged, due to some safeguards built into the  system.   However,  you 
    should  immediately  shift-reset  and reboot, and not try to warm-start 
    the  system,  since  the  CPU  may  have   damaged  tables  in  memory. 
    Remember, the HDOS system uses a sophisticated linked-allocation scheme 
    to  handle disk files.  Damaging that table or damaging  the  directory 
    or  allocation  areas  on  the disk can cause all files on that disk to 
    become lost; not just one or two!             
     
    If  you  are  debugging  a  program  which  consistently  vectors  into 
    undefined memory locations, then it is best to use write-protect labels 
    on  the  disks.   Then  if  you crash, you can quickly restart by using 
    PAM-8 to start at the HDOS cold start address,  040100A.   Entering  at 
    this  address  should return you to HDOS command mode.  Do this only if 
    you have your disks write-protected.  Otherwise, it is too risky. 

                         PART 4 - PRECAUTIONS (Cont) 
                         +++++++++++++++++++++++++++ 
 
    CPU PRECAUTIONS (Cont) 
    ====================== 
 
    Usually,  when  your  program  runs  wild, the CPU ends up at some high 
    memory location where you don't have any memory.  The computer hardware 
    generates  0  for nonexistent memory, so you will quickly run through a 
    long  string of NOPs until you wrap from 377377A to 000000A,  which  is 
    the  master clear restart address for PAM-8.  If you display the PC and 
    find it set to your high memory address, then you  probably  took  this 
    "circumpolar" route into PAM-8.             
     
 
    Debugging Hints 
    =============== 
 
    The best way to debug programs is to ORG them above DBUG, and test them 
    using DBUG.  After entering DBUG, use the LOAD command to load  in  the 
    program  under  test.  You can then break-point and single-step through 
    your program.  Do not single-step through an HDOS  SCALL,  or  you  may 
    damage the disk.            
     
    After  the program seems to be working, ORG it back down to 042200A, or 
    wherever, and reassemble.            
    *********************************************************************** 

                           PART 5 - RESIDENT SCALLS 
                           ++++++++++++++++++++++++ 
 
    INTRODUCTION 
    ============ 
 
    Within HDOS 3.0 all system calls are "Resident Scalls," since there are 
    no overlays.  This  segment  covers  those  HDOS  system  calls,  often 
    referred to as "SCALLS," which are permanently resident in memory. 
 
    In general for the H8, a SCALL (System CALL) consists of a 
 
            RST    7 
                                              
    instruction  followed  by  a  byte containing the request number.  Most 
    SCALLs  require  that  some  registers  be  set  up  before  the  call. 
    Likewise,  most  may  alter the registers, so a program should save any 
    registers which it wants to preserve.             
     
    The  ASM  (assembler)  has  a special opcode for SCALLs: 'SCALL code' - 
    where "code" is the number of the request.   This  statement  generates 
    the equivalent of: 
 
                       DB    377Q,code 
 
    We  recommend  that  you  use  the HOSDEF.ACM file to include all these 
    definitions.  In general, it is advisable to use the recommended symbol 
    definitions for all references to HDOS, and include them in one or more 
    XTEXT decks.  This will make programming easier for you, and  guarantee 
    compatibility  with  future HDOS releases.  Although we will make every 
    effort to keep binary compatibility, we may need to revert to "assembly 
    language compatibility," in which case you may have to change some HDOS 
    symbol values and reassemble.             
    ....................................................................... 
    The SCALLs for HDOS Versions 3.0/3.02 are listed in numerical order: 
    ....................................................................... 
 
    .EXIT - Exit User Program (Octal 0Q) 
    ==================================== 
 
    ***     EXIT - EXIT USER PROGRAM            
    *       
    *       EXIT IS CALLED TO RETURN CONTROL TO THE SYSTEM COMMAND 
    *       PROGRAM.           
    * 
    *       FOR A NORMAL EXIT, THE CONTROL CHARACTERS ARE  
    *       CLEARED, AND SYSCMD IS ENTERED.     
    * 
    *       MVI       A,FLAG          (see below) 
    *       SCALL     .EXIT 
    * 
    *       FOR EITHER EXIT, THE CONTROL CHARACTER VECTORS 
    *       (SET BY .CTLC) ARE CLEARED.         

                           PART 5 - RESIDENT SCALLS 
                           ++++++++++++++++++++++++ 
 
    .EXIT - Exit User Program (Octal 0Q) (Cont) 
    =========================================== 
 
    * 
    *       IN ADDITION, THE ABORT EXIT RESETS THE DISK AND 
    *       CONSOLE I/O DRIVERS.        
    * 
    *       ENTRY     (A)       = FLAG ( 0 = NORMAL, 1 = ABORT ) 
    *       EXIT      -IF-      [ SYSTEM DISK IS STILL MOUNTED ] 
    *                                  - or - 
    *                           [ STAND-ALONE IS SET ] 
    *                 -THEN-    EXIT TO "SYSCMD.SYS" 
    *                 -ELSE-    EXIT TO REBOOT CODE 
 
 
    Notes: 
    ------ 
    The  .EXIT  SCALL  is the proper way for a program to return control to 
    HDOS.  In any mode, .EXIT will  close  all  open  I/O  channels.   This 
    action  is  equivalent  to  that of the .CLEAR SCALL.  It is best for a 
    program to close or clear its own channels before incurring  .EXIT,  as 
    future releases may differ in this action.            
     
    It  should  not be necessary for a program to use the abort exit unless 
    some process was being used which affected the state of the console  or 
    disk I/O ports.  The use of such processes is not recommended.  
     
    If  SY0: has been dismounted, and the STAND-ALONE flag is not set, HDOS 
    exits to reboot.  If the STAND-ALONE flag has been set, and no disk  is 
    mounted  on  SY0:,  or  SYSCMD.SYS  is not found on the disk mounted on 
    SY0:, HDOS  exits to reboot.  Thus, the  only  way  for  a  program  to 
    return to the command level once SY0: has been dismounted and remounted 
    is for the STAND-ALONE flag to have  been previously set  via  the  SET 
    command,  and for the disk mounted on SY0: to have a copy of SYSCMD.SYS 
    on it.    
     
 
    **      EXAMPLES: 
    ALDONE  MVI    A,0          FLAG NORMAL EXIT 
            SCALL  .EXIT 
    ABTXIT  MVI    A,1          FLAG ABORT EXIT 
            SCALL  .EXIT 

                           ++++++++++++++++++++++++ 
 
    .EXIT - Exit User Program (OCTAL 0Q)(Cont) 
    ========================================== 
 
                                     NOTE 
              You  should  always close your open channels before 
              exiting.  Do not rely upon .EXIT to do it for  you. 
              Future  HDOS  releases may not be so accommodating. 
              The one exception is for temporary  scratch  files, 
              for  which you may use .CLEAR.  This will cause the 
              file  to be dissolved and its space returned to the 
              free pool. 
    ....................................................................... 
 
    .SCIN - System Console Input (Octal 1Q) 
    ======================================= 
 
    ***     SCIN - SYSTEM CONSOLE INPUT. 
    * 
    *       .SCIN TAKES A SINGLE CHARACTER FROM THE CONSOLE INPUT 
    *       BUFFER, IF ANY ARE AVAILABLE. 
    *        
    *  L1   SCALL   .SCIN 
    *       JC      L1          CHARACTER NOT READY 
    * 
    *       ENTRY   NONE 
    *       EXIT    'C' SET IF NO CHARACTER 
    *               'C' CLEAR IF CHARACTER 
    *               (A) =CHARACTER 
    *       USES    A,F 
 
    Notes: 
    ------ 
    This  command  is  relatively  obvious.  Detailed examples of .SCIN are 
    shown in the HEATH HDOS Software Reference Manual, Chapter 11, Appendix 
    11-B, pages 11-70 thru 11-76.  A simple application example follows: 
     
 
    **      EXAMPLE: 
 
    RDCHAR  SCALL   .SCIN       TRY TO READ CHARACTER 
            JC      RDCHAR      NONE READY YET 
            RET                  EXIT, (A) = CHARACTER 
 
        
                                     NOTE 
          The .CONSL SCALL may be used to set console mode bits. 
    ....................................................................... 

                           PART 5 - RESIDENT SCALLS 
                           ++++++++++++++++++++++++ 
 
    .SCOUT - System Console Output (Octal 2Q) 
    ========================================= 
 
    ***     SCOUT - SYSTEM CONSOLE OUTPUT 
    * 
    *       SCOUT OUTPUTS A SINGLE CHARACTER TO THE CONSOLE.  CURSOR 
    *       POSITIONING IS KEPT TRACK OF.  A 'NL' CHARACTER 
    *       INDICATES A NEW LINE.  'CR' AND 'LF' CHARACTERS 
    *       SHOULD NOT BE USED. 
    * 
    *       MVI     A,CHAR 
    *       SCALL   .SCOUT 
    * 
    *       ENTRY   (A) = CHARACTER 
    *       EXIT    (A) = CHARACTER 
    *       USES    NONE 
 
    Notes: 
    ------ 
    This  command  is  relatively obvious.  Detailed examples of .SCOUT are 
    shown in the HEATH HDOS Software Reference Manual, Chapter 11, Appendix 
    11-B, pages 11-71 through 11-76.  A simple application example follows: 
     
 
    **      EXAMPLE:  
 
            MVI     A,'*' 
            SCALL   .SCOUT      TYPE AN ASTERISK ON THE CONSOLE 
 
    ....................................................................... 
 
 
    .PRINT - Print Line On System Console (Octal 3Q) 
    ================================================ 
 
    ***     .PRINT - PRINT CONSOLE LINE. 
    * 
    *       PRINT CAUSES A CODED LINE TO BE PRINTED AT THE CONSOLE. 
    * 
    *       LXI     H,LINEADDR 
    *       SCALL   .PRINT 
    * 
    *       THE LAST CHARACTER IN THE LINE SHOULD HAVE THE 
    *       200Q BIT SET. 
    *        

                           PART 5 - RESIDENT SCALLS 
                           ++++++++++++++++++++++++ 
 
    .PRINT - Print Line On System Console (Octal 3Q) (Cont) 
    ======================================================= 
 
    * 
    *       ENTRY   (HL) = LINE ADDRESS 
    *       EXIT    (HL) = LWA OF MESSAGE +1 
    *       USES    A,F,H,L 
 
 
    Notes: 
    ------ 
    .PRINT  is an efficient and convenient way to print lines on the system 
    console.  Another good way is to use the subroutine "$TYPTX," as  shown 
    in  Part  8,  Appendix  13-A, pages 13-73 through 13-76 of this manual. 
    Note that the parity bit (200Q) is set over the last  character  to  be 
    printed  to  notify  the  end-of-line  to  HDOS.   Remember, use the NL 
    character  (012Q,  same  as  LF)  for  a  CRLF  sequence.   HDOS   will 
    automatically  insert  the  required  number  of PAD characters for the 
    console.  If you prefer, you can include the "NULL 00" character  in  a 
    print  line.   It is ignored, does not cause a delay in console output, 
    and thus cannot be used as a PAD character. 
     
     
    ***     EXAMPLE: 
            . 
            LXI     H,MSGA      TYPE OUT STARTUP MESSAGE 
            SCALL   .PRINT 
            . 
            . 
    PROMPT  LXI     H,MSGB      TYPE OUT PROMPT MESSAGE 
            SCALL   .PRINT 
    REACHA  SCALL   .SCIN       READ REPLY.... 
            . 
            . 
    MSGA    DB      12Q,'SET OPTIONS:' 
            DB      12Q 
            DB      12Q, 'HELP - TYPE THIS LIST' 
            DB      12Q, 'CRASH - DESTROY DISK SURFACE' 
            DB      12Q+200Q                   NEW LINE, END OF PRINT 
           
    MSGB    DB      12Q,'YOUR COMMAND?',' '200Q 
    ....................................................................... 

    .READ - Read From File (Octal 4Q) 
    ================================= 
 
    Notes: 
    ------ 
    Use  the  .READ  SCALL  to read data from an open channel.  The channel 
    must already have been opened via a .OPENR or .OPENU SCALL (except  for 
    channel -1, as noted previously).              
     
    Currently,  all  device  I/O  under  HDOS  (with  the  exception of the 
    console, via the .SCIN and .SCOUT SCALLs) is "block mode."  This  means 
    that  you  must  read or write to the device in multiples of 256 bytes. 
    If you cannot fill in the last block,  you  should  pad  it  with  zero 
    bytes.   The  last  block in all HDOS source files is padded out to 256 
    characters with 00 bytes.            
     
    The  quoted  C in the following example indicates the Carry Flag.  This 
    SCALL, as in all others in HDOS, returns with the carry flag set if  an 
    error  or abnormal condition has occurred.  The most common "error" for 
    the .READ command is "end-of-file."   The  convention  used  above  and 
    throughout  this manual is that exit conditions which are predicated on 
    the setting of a flag are discussed directly under that flag,  indented 
    one space.  Thus, the (BC) register pair contains the unused byte count 
    if, and only if, the "C" flag is set.  If "C" is clear, then all of the 
    bytes  were  read,  and (BC) contains garbage.  Thus, the (BC) and (DE) 
    registers contain meaningful information only when an  error  condition 
    occurred, which is normally an "end-of-file."  The error codes returned 
    by HDOS are defined in Part 7, pages 13-67 and 13-68  of  this  manual. 
    This  is  simply  a condensation of the error messages discussed in the 
    HDOS Software Reference Manual.  (For details  concerning  HDOS  system 
    error  messages,  refer  to Chapter 11, pages 11-61.  Note that you can 
    use the .ERROR SCALL to look up an explanatory message. 
     
     
    ***     .READ - BLOCKS OF DATA 
    * 
    *       READ PROCESSES READ SCALLS.  IF A SERIAL DEVICE, PASS TO 
    *       DRIVER.  IF A STORAGE DEVICE, HANDLE STORAGE MAPPING. 
    * 
    *       MVI     A,CHAN 
    *       LXI     B,COUNT     MUST BE MULTIPLE OF 256 
    *       LXI     D,ADDR 
    *       SCALL   .READ       READ DATA FROM FILE 
    * 
    *       ENTRY   (A) = I/O CHANNEL/NUMBER 
    *               (B) = COUNT OF 256-BYTE BLOCKS TO TRANSFER 
    *               (C) = 0 
    *               (DE) = DATA ADDRESS 
    * 

    .READ - Blocks Of Data (Octal 4Q)(Cont) 
    ======================================= 
 
    * 
    *       EXIT    'C' CLEAR IF ALL OK 
    *               'C' SET IF ERROR 
    *               (A) = ERROR CODE 
    *               (BC) = UNUSED TRANSFER COUNT 
    *               (DE) = NEXT UNUSED ADDRESS 
    *       USES    ALL 
 
 
                                     NOTE 
              All  read  operations must be for integer multiples 
              of 256 bytes.  Thus, the last sector in a file  may 
              have to be padded with 00 bytes.  All ASCII (coded) 
              files in HDOS are  zero-byte  filled  in  the  last 
              sector  (if they need it).  A 00 byte is considered 
              to  be  a  NULL  character  and  should  always  be 
              ignored when encountered in an ASCII file.  
               
    ***     EXAMPLES: 
 
            . 
            . 
    READ    MVI     A,1         READ FROM ALREADY OPEN CHANNEL 1 
            LXI     B,256       READ ONE SECTOR 
            LXI     D,BUFFER     
            SCALL   .READ       READ IT 
            JC      READ1       ERROR 
            LXI     B,256       READ 256 BYTES 
            JMP     READ2 
 
    *       HAVE ERROR.  SEE IF EOF, OR SOMETHING WORSE  
    READ1   CPI     EC,EOF      SEE IF JUST EOF 
            JNE     ERROR       HAVE SERIOUS ERROR 
            STA     EOFFLG      FLAG HAVE SEEN EOF 
            LXI     H,256       (HL) = ORIGINAL STARTING COUNT 
            MOV     A,L 
            SUB     C 
            MOV     C,A 
            MOV     A,H 
            SBB     B 
            MOV     B,A         (BC) = 256-REMCNT = AMOUNT READ 
 
    *       READ COMPLETE.  (BC) = BYTES AVAILABLE 
    READ2   .... 
            . 
            . 
    BUFFER  DS      256         SECTOR BUFFER 
    ....................................................................... 

    .WRITE - Write to Open File (Octal 5Q) 
    ====================================== 
 
    ***     .WRITE - PROCESS WRITE SCALL. 
    * 
    *       MVI     A,CHAN 
    *       LXI     B,COUNT     MUST BE MULTIPLE OF 256 
    *       LXI     D,ADDR 
    *       SCALL   .WRITE      WRITE DATA TO CHANNEL 
    * 
    *       ENTRY   (A) = CHANNEL # 
    *               (BC) = DATA COUNT 
    *               (DE) = DATA ADDRESS 
    *       EXIT    'C' CLEAR IF ALL OK 
    *               'C' SET IF ERROR 
    *                (BC) = UNUSED TRANSFER COUNT 
    *                (DE) = NEXT UNUSED ADDRESS 
    *                (A) = ERROR CODE 
    *       USES     ALL 
 
    Notes: 
    ------ 
    The  .WRITE  SCALL  is  very similar to the .READ SCALL, except that it 
    writes the data to the file.  Once again, the count in (BC) must be  an 
    integral multiple of 256.  The most typical error returned by .WRITE is 
    "NO ROOM ON MEDIA."             
     
                                     NOTE 
              All  write operations must be for integer multiples 
              of 256 bytes.  Thus, the last sector in a file  may 
              have  to  be  filled  out  to 256 bytes.  All ASCII 
              (coded) files in HDOS are zero-byte filled  in  the 
              last  sector  (if  they  need  it).   A  00 byte is 
              considered  a NULL character and should  always  be 
              ignored when encountered in an ASCII file.           
               
 
    **      EXAMPLE: 
 
    WRIDAT  MVI     A,1         CHANNEL 1 ALREADY OPEN 
            LXI     B,512       WRITE 512 BYTES 
            LXI     D,BUFFER     
            SCALL   .WRITE      WRITE IT 
            JC      ERROR       SERIOUS ERROR 
            . 
            . 
    BUFFER  DS      512         BUFFER AREA FOR WRITE 
    ....................................................................... 

    .CONSL - Set Console Mode Bits (Octal 6Q) 
    ========================================= 
 
    ***     .CONSL - SET AND CLEAR CONSOLE FLAGS. 
    * 
    *       CONSL IS CALLED TO SET, CLEAR, OR READ BITS IN THE 
    *       VARIOUS CONSOLE FLAGS. 
    * 
    *       THE CALLER PASSES AN INDEX INTO THE PROPER FLAG.  A 
    *       MASK TO INDICATE THE AFFECTED BITS, AND A SET OF NEW 
    *       VALUES FOR THOSE BITS. 
    *            
    *       INDEX = 
    * 
    *       0       I.CSLMD 
    *       1       I.CONTY 
    *       2       I.CUSOR 
    *       3       I.CONWI 
    *       4       I.CONFL 
    * 
    *       ENTRY   (A) = INDEX 
    *               (B) = NEW VALUES 
    *               (C) = MASK ('1' BIT FOR EVERY BIT TO CHANGE) 
    *       EXIT    'C' CLEAR IF NO ERROR 
    *                (A) = NEW VALUE 
    *               'C' SET IF ERROR 
    *                (A) = ERROR CODE 
    *       USES    ALL 
 
 
    Notes: 
    ------ 
    The .CONSL SCALL is used to read and write the console control bits and 
    bytes.  These bytes are available directly in memory, but we  recommend 
    that   you   access   them   via   the   .CONSL  command  to  guarantee 
    synchronization and upward compatibility with future releases of HDOS. 
 
    The caller supplies HDOS with three values: the index of the byte to be 
    read and/or written, the bits to be altered, and the  new  bit  values. 
    The  technique  of  supplying  a "bits-affected" mask and a "new value" 
    pattern  allows you to alter just one bit in a byte without  having  to 
    know the values of the other bits in the byte.  Since the console is an 
    interrupt-responsive device, this also avoids synchronization problems. 
    There  are  five  bytes which may be read and/or written via the .CONSL 
    function.            

    .CONSL - Set Console Mode Bits (Octal 6Q)(Cont) 
    ================================================ 
 
    I.CSLMD - Console Mode 
 
    I.CSLMD EQU     0           I.CSLMD IS FIRST BYTE 
 
    CSL.ECH EQU     10000000B   SUPPRESS ECHO 
    CSL.WRP EQU     00000010B   WRAP LINES AT WIDTH 
    CSL.CHR EQU     00000001B   UPDATE IN CHARACTER MODE 
 
    These  three  bits  are  used  to affect the mode in which HDOS handles 
    characters typed at the console.  They are documented in greater detail 
    in  the  "HDOS  Software  Reference  Manual,"  Chapter  11, pages 11-70 
    through 11-76. 
     
    I.CONTY - Console Type 
    ---------------------- 
    I.CONTY EQU     1           I.CONTY IS 2ND BYTE 
 
    CTP.BKS EQU     10000000B   TERMINAL PROCESSES BACKSPACES 
    CTP.MLI EQU     00100000B   MAP LOWER CASE TO UPPER ON INPUT 
    CTP.MLO EQU     00010000B   MAP LOWER CASE TO UPPER ON OUTPUT 
    CTP.2SB EQU     00001000B   TERMINAL NEEDS TWO STOP BITS 
    CTP.BKM EQU     00000010B   MAP BKSP (UPON INPUT) TO RUBOUT 
    CTP.TAB EQU     00000001B   TERMINAL SUPPORTS TAB CHARACTERS 
 
    The  bits  in  the  I.CONTY  byte  are  used  to describe the console's 
    hardware characteristics.  These bits are all discussed under  the  SET 
    command  section  in the "HDOS Software Reference Manual."  See Chapter 
    3, page 3-19, for details.            
 
    I.CUSOR - Console Cursor Position 
    --------------------------------- 
    I.CUSOR EQU     2           I.CUSOR IS 3RD BYTE 
 
    The I.CUSOR  byte  contains  the current cursor position of the console 
    terminal cursor.  Immediately after a NEW LINE, this byte contains 001. 
                
    I.CONWI - Console Width 
    ----------------------- 
    I.CONWI EQU     3           I.CONWI IS 4TH BYTE 
 
    The  I.CONWI  byte  contains  the current console width.  This value is 
    documented under the SET command in the HDOS Software Reference Manual. 
    See  Chapter  3,  page  3-19, for  details.   In brief, when the cursor 
    reaches this value, HDOS automatically generates a NEW LINE.   You  can 
    effectively disable this option by setting the width to 255. 

    .CONSL - Set Console Mode Bits (Octal 6Q)(Cont) 
    =============================================== 
 
    I.CONFL - Console Flags 
    ----------------------- 
    I.CONFL EQU     4           I.CONFL IS 5TH BYTE 
 
    CO.FLG  EQU     00000001B   CTL-O FLAG 
    CO.FLG  EQU     10000000B   CTL-S FLAG 
 
    The  I.CONFL byte contains the current setting of the console CTL-O and 
    CTL-S bytes.  A user program may find it useful to note that  the  user 
    has  typed CTL-O or CTL-S.  In addition, your program may want to clear 
    the CTL-O flag immediately before an input prompt is typed, so that the 
    typing of the prompt is guaranteed.             
                                 
                                     NOTE 
              If the CTL-S flag is set, and your program issues a 
              character to the console via .SCOUT or .PRINT, then 
              your  program will hang up in HDOS, waiting for the 
              CTL-S flag to clear.  There  is  no  way  to  do  a 
              "conditional"  character type out.  Programmers who 
              do not want their programs to hang  up  must  check 
              the  status of the CTL-S flag after every .SCOUT or 
              .PRINT, and trust to luck  that  the  user  doesn't 
              type the CTL-S between the .CONSL and the .SCOUT. 
                          
 
    **      EXAMPLES: 
 
    *       SET CHARACTER MODE, NO ECHO 
      
            MVI     A,I.CSLMD   (A) = BYTE INDEX 
            MVI     B,CSL.ECH+CSL.CHR   SET BOTH BITS 
            MVI     C.CSL.ECH+CSL.CHR   AFFECT BOTH BITS 
            SCALL   .CONSL 
 
    *       SET MAP LOWER CASE TO UPPER, CLEAR BACKSPACE ON 'RUBOUT' KEY 
 
            MVI     A,I.CONTY   (A) = BYTE INDEX 
            MVI     B,CTP.MLI+CTP.MLO   SET MAP LOWER CASE BITS 
            MVI     C,CTP.MLI+CTP.MLO+CTP.BKS   SET MAP, CLEAR BKS 
            SCALL   .CONSL 
 
    *       READ CONSOLE CURSOR POSITION 
            MVI     A,I.CUSOR 
            MVI     C,O         AFFECT NO BITS, (B) MEANINGLESS 
            SCALL   .CONSL      AFFECT NOTHING, JUST GET NEW 
                                (SAME AS OLD) VALUE 
            CPI     1           SEE IF CURSOR OVER COLUMN 1 

    .CONSL - Set Console Mode Bits (Octal 6Q)(Cont) 
    =============================================== 
 
    *       SET CONSOLE WIDTH 
 
            MVI     A,I.CONWI 
            MVI     B,80        SET 80 COLUMNS 
            MVI     C,377Q      AFFECT FULL BYTE 
            SCALL   .CONSL      SET WIDTH 
    ....................................................................... 
 
 
    .CLRCO - Clear Console Buffer (Octal 7Q) 
    ======================================== 
 
    ***     .CLRCO - CLEARS THE CONSOLE BUFFERS. 
    * 
    *       CLRCO CLEARS THE CONSOLE TYPE-AHEAD BUFFER. 
    *       CTL-O AND CTL-S FLAGS ARE ALSO CLEARED. 
    * 
    *       EMTRY   NONE 
    *       EXIT    NONE 
    *       USES    ALL 
 
 
    Notes: 
    ------ 
    The  .CLRCO  SCALL is  used to clear the console buffer and the console 
    CTL-O and CTL-S flags.  HDOS contains a  "type-ahead"  buffer,  so  the 
    user  may  type  commands  before  a program asks to read them from the 
    console.  All typed text is stored in  this  "type-ahead"  buffer;  the 
    .SCIN  SCALL reads the characters from the buffer.  The special control 
    characters, CTL-A, CTL-B, and CTL-C, are not stored  in  the type-ahead 
    buffer,  but  instead  cause an interrupt to a user service routine (if 
    you set one up via the .CTLC SCALL).  Often a user has typed a  partial 
    line  before  he has typed the CTL-C (or CTL-A, or CTL-B).  You can use 
    the .CLRCO function to clear out any unwanted type-ahead characters. 
               
 
                                     NOTE 
              Issuing  the  .CLRCO  function does not cause a NEW 
              LINE to be sent to the console.  The user is  given 
              no indication that the characters he may have typed 
              in have been discarded.  Your program should  issue 
              a  new prompt immediately after the .CLRCO function  
              to make things clear to the user.              
 
 
 
 
 
 



    CHAPTER THIRTEEN    HDOS PROGRAMMERS' REFERENCE MANUAL       PAGE 13-25 
    ================    ==================================       ========== 
 
 
                           PART 5 - RESIDENT SCALLS 
                           ++++++++++++++++++++++++ 
 
    .CLRCO - Clear Console Buffer (Octal 7Q)(Cont) 
    ============================================== 
 
    ***     EXAMPLE: CLEANUP AFTER CTL-C 
 
    *       ASSUME CONTROL PASSES HERE AT CTL-C 
 
    CCHIT   LXI     H,CCHITA    TYPE CTL-C 
            SCALL   .PRINT      ACKNOWLEDGE CTL-C, SETUP NEW LINE 
            SCALL   .CLRCO      CLEAR TYPE-AHEAD 
            . 
            . 
    CCHITA  DB      '^C',212Q   ^C WITH NEW-LINE 
 
 
 
                                     NOTE 
                   PART 6 discusses intercepting CTL-Cs.    
    ....................................................................... 

    .VERS - HDOS Version Number (Octal 11Q) 
    ======================================= 
 
    ***     VERS - RETURN HDOS VERSION NUMBER 
    * 
    *       VERS RETURNS THE HDOS VERSION NUMBER AS A ONE-BYTE 
    *       BCD NUMBER.  A DECIMAL IS ASSUMED BETWEEN THE HIGH 
    *       AND LOW ORDER NYBBLES. 
    *        
    *       ENTRY   NONE 
    *       EXIT    (PSW)       = 'C' CLEAR IF NO ERROR 
    *                              (A) = VERSION NUMBER 
    *                             'C' SET IF ERROR ( VERS < 1.5 ) 
    *                              (A) = ERROR CODE ( EC.ILC ) 
    * 
    *       USES    A,F 
 
 
    Notes: 
    ------ 
    The  .VERS system call returns the current version number of HDOS.  The 
    primary use of this system call is to ascertain under which version  of 
    HDOS  the  program  is  running.   If  the  program determines that the 
    version does not support these new SCALLs, it may exit gracefully  with 
    an  error  message.                                                     
     
    The  version  number is returned as one BCD byte.  That is, version 1.5 
    will return 21, 25Q, or 015H.  Refer to the HDOS  Common  Deck  listing 
    for an example of the definition format. 
     
     
    ***     EXAMPLE: 
 
            SCALL   .VERS 
            JC      BADVER      No version system call 
            CPI     VERS 
            JNZ     BADVER      Invalid version 
            . 
            . 
    BADVER  LXI     B,MESSAG 
            SCALL   .PRINT 
            . 
            . 
    MESSAG  DB      12Q,'This Version of HDOS Does Not Support' 
            DB      'the Required System Calls.',12Q+200Q 
    ....................................................................... 

    .GDA - Get Device Driver Address (Octal 12Q) 
    ============================================ 
 
    ***     GDA - Get Device Driver Address. 
    * 
    *       Entry:  (DE) = Device Name 
    *       Exit:   'C' Clear 
    *                (HL) = Driver address 
    *                (BC) = Table address for this device 
    *               'C' Set if error 
    *                (A) = Error Code 
    *                (HL) = (BC) = 0 
    *       Uses:   A,F,H,L,B,C 
 
    ....................................................................... 
 
    .CRC16 - CRC-16 Is A Block Of Memory (Octal 13Q) 
    ================================================ 
 
    ***     CRC16 - CRC-16 A Block of memory. 
    * 
    *       Entry:  (HL) = BUFFER ADDRESS 
    *               (DE) = INITIAL CRC-16 
    *               (BC) = LENGTH OF DATA 
    *       Exit:   (HL) = ADVANCED PAST BUFFER 
    *               (DE) = UPDATED CRC-16 
    *               (BC) = 0 
    *       Uses:   ALL 
 
    ....................................................................... 
 
 
    .LINK - Process Link Scall (Octal 40Q) 
    ====================================== 
                  
    ***     .LINK - PROCESS LINK SCALL 
    * 
    *       LINK LOADS IN AND RUNS ANOTHER PROGRAM.  THE OPEN FILES 
    *       SYSTEM TABLES, AND STACK ARE NOT DISTURBED. 
    * 
    *       ENTRY   (HL) = ADDRESS OF PROGRAM FILE DESCRIPTOR 
    *       EXIT    TO LINKED PROGRAM, IF OK 
    *                (A) UNCHANGED 
    *                (SP) = VALUE AT 'LINK' SCALL 
    *               TO CALLER IF ERROR 
    *                'C' SET 
    *                (A) = ERROR CODE 
    *       USES    ALL 

    .LINK - Process Link Scall (Octal 40)(Cont) 
    =========================================== 
 
    Notes: 
    ------ 
    The .LINK SCALL is used to pass control to another program.  
 
    **      EXAMPLE: TRANSFER CONTROL TO PROGRAM 'CLEANUP.ABS' 
 
    XFER    MVI     A,-1        CHANNEL -1 OPEN ON LOADED FILE 
 
    *       GET DEVICE WERE LOADED FROM, SO THAT WE CAN  
    *       RUN 'CLEANUP.ABS' FROM THAT SAME DISK 
 
            LXI     D,DEVCODE   AREA FOR DEVCODE 
            LXI     H,BUFFER    PUT NAME INTO SCRATCH AREA 
            SCALL   .NAME 
 
    *       BUILD NAME TO LINK TO ... 
 
            LXI     B,XFERAL    (BC) = NUMBER OF BYTES TO MOVE 
            LXI     D,XFERA      FROM XFERA 
            LXI     H,DEVCODE+3  PUT AFTER DEVICE SPECIFICATION 
            CALL    $MOVE        PUT NAME AFTER DEVICE (ROUTINE 
                                 IN H17 ROM) 
 
    *       CALL PROGRAM 
 
            LXI     H,DEVCODE 
            SCALL   .LINK       TRY TO EXECUTE IT 
            JC      ERROR       FAILED 
 
    XFERA   DB      ':CLEANUP.ABS',0     NAME 
    XFERAL  EQU     *-XFERA     AMOUNT TO MOVE 
 
    DEVCODE DS      3+XFERAL    ROOM FOR ENTIRE FILE SPECIFICATION 
    ....................................................................... 

    .CTLC - Set Up Handlers for Control Characters (Octal 41Q) 
    ========================================================== 
 
    ***     .CTLC - SET CONTROL CHARACTER ADDRESS 
    * 
    *       THE .CTLC SCALL IS USED TO SET UP HANDLING FOR 
    *       THE CONTROL CHARACTERS CTL-A, CTL-B, AND CTL-C. 
    * 
    *       A SEPARATE ADDRESS IS SPECIFIABLE FOR EACH CHARACTER. 
    *       IF AN ADDRESS OF 0 IS SPECIFIED, PROCESSING OF THAT 
    *       CHARACTER IS SUSPENDED. 
    * 
    *       THE PROCESS ADDRESS MUST BE >255A. 
    * 
    *       ENTRY   (A) = CONTROL CHARACTER WHOSE PROCESS ADDRESS IS 
    *                     TO CHANGE (CTL-A, CTL-B, OR CTL-C) 
    *               (HL) = NEW ADDRESS (=0 TO CLEAR PROCESSING) 
    *       EXIT    'C' CLEAR IF OK 
    *               'C' SET IF ERROR 
    *                (A) = ERROR CODE 
    *       USES    A,F,H,L 
 
    Notes: 
    ------ 
    The .CTLC SCALL allows you to set up interrupt service routines for the 
    handling of CTL-A, CTL-B, and CTL-C.  You may set up a separate service 
    routine for each character.              
     
    When a service routine has been set up, and the specified character has 
    been struck, your routine  will  be  entered  at  interrupt-time,  with 
    interrupts enabled. 
     
    Upon  entry  to  your  routine,  the  registers  B, C, D, E, and L have 
    whatever contents were in them at the time  of  the  control  character 
    interrupt.  The stack contains: 
     
                ((SP)+0) = Return Address into HDOS 
                ((SP)+2) = Interrupted PSW 
                ((SP)+4) = Interrupted PC 
 
    Your  routine  can  do  some  interrupt-time  work  (having  saved  the 
    registers first, of course) and then do a RET to HDOS,  in  which  case 
    HDOS  will  take care of the rest.  Or, if you wish, you may ignore the 
    HDOS  return  address and jump back into your program's command loop or 
    whatever.            

    .CTLC - Set Up Handlers For Control Characters (Octal 41Q)(Cont) 
    ================================================================ 
 
    **      EXAMPLE 1: SETTING AN 'INTERRUPT OCCURRED' FLAG 
 
            LXI     H,CCINT     SET UP CTL-C INTERRUPT PROCESSOR 
            MVI     A,003       (A) = CTLC 
            SCALL   .CTLC       SET UP CTL-C 
            . 
            . 
    LOOP    SCALL   .SCIN 
            JNC     GOTONE      GOT A CHARACTER 
            LDA     CCHIT 
            ANA     A 
            JZ      LOOP        NO CTL-C HIT 
            JMP     PROCC       PROCESS CTL-C 
            . 
            . 
    *       CTL-C CAUSES THIS ROUTINE TO BE ACTIVATED 
 
    CCINT   MVI     A,1         PSW IS ALREADY SAVED 
            STA     CCHIT       SET CC HIT 
            RET                 RETURN TO INTERRUPTED CODE VIA  
    HDOS 
            . 
            . 
    CCHIT   DB      0           SET =1 WHEN CTL-C TYPED 
 
 
    **      EXAMPLE 2: RETURNING CONTROL TO MAIN COMMAND LOOP. 
 
            . 
 
            . 
            . 
            LXI     H,CBHIT 
            MVI     A,002       (A) = CTLB 
            SCALL   .CTLB 
            . 
            . 
    START   LXI     SP,STACK    CLEANUP STACK 
    LOOP    .                   DO WHATEVER WE DO... 
            . 
            . 
    *       ENTERED HERE IF CTL-B HIT 
 
    CBHIT   JMP     START       RESTART COMMAND LOOP 
    ....................................................................... 

    .OPENR - Open File for Read (Octal 42Q) 
    ======================================= 
 
    ***     .OPENR - OPENR SCALL PROCESSOR. 
    * 
    *       OPENR IS CALLED TO OPEN A CHANNEL FOR READ. 
    * 
    *       THE CALLER SUPPLIES A FILE NAME.  A DEFAULT BLOCK 
    *       FOR THE DEVICE AND EXTENSION, AND A CHANNEL NUMBER. 
    * 
    *       DEFAULT BLOCK FORMAT: 
    * 
    *       DB      'DDD'       DEFAULT DEVICE 
    *       DB      'XXX'       DEFAULT EXTENSION 
    * 
    *       ENTRY   (DE) = DEFAULT BLOCK ADDRESS 
    *               (HL) = NAME ADDRESS 
    *               (A) = CHANNEL NUMBER 
    *       EXIT    'C' CLEAR IF OK 
    *               (HL) = ADVANCED PAST FILE NAME 
    *               'C' SET IF ERROR 
    *                (A) = ERROR CODE 
    *       USES    ALL 
      
 
    Notes: 
    ------ 
    Use  the  .OPENR  SCALL to open files for read access.  This means that 
    you may then read the file, but HDOS will not allow any write  requests 
    to  it.   You  may  open  an individual file for read access on as many 
    channels as you wish.           
     
    The  channel  number supplied must be a legal one (i.e., -1 through 5), 
    and must not already have a file open on it.            
     
    HDOS  will not allow any one file to be open for both read and write at 
    the same time, nor may any one file be open for write to more than  one 
    channel.   Attempting  to  do this will cause a "usage conflict" error. 
    This means that you may not open a file via .OPENR  if  it  is  already 
    open for write, or update, on another channel.             
 
    **      EXAMPLE: 
 
    *       OPEN PRE-DETERMINED FILE NAME ON CHANNEL 1 
 
            MVI     A,1         CHANNEL 1 
            LXI     D,DEFALT    POINT TO DEFAULT BLOCK 
            LXI     H,FNAME     POINT TO FILE DESCRIPTOR 
            SCALL   .OPENR      OPEN FOR READ 
            JC      ERROR       SOME ERROR 

    .OPENR - Open File For Read (Octal 42Q)(Cont) 
    ============================================= 
 
            . 
            . 
 
    *       READ FILE NAME FROM USER, OPEN ON CHANNEL 2 
 
            LXI     H,MSGA 
            SCALL   .PRINT      PROMPT HIM 
            LXI     H,BUFFER 
    REA1    SCALL   .SCIN 
            JC      REA1        NO CHARACTER 
            MOV     M,A         STORE IN MEMORY 
            INX     H 
            CPI     012Q        SEE IF NEW LINE (USER HIT CR KEY) 
            JNE     REA1        NOT YET 
            DCX     H                                         
            MVI     M,0         TERMINATE LINE WITH 00, INSTEAD 
                                OF 012Q 
 
            LXI     H,BUFFER 
            LXI     D,DEFALT    POINT TO DEFAULT BLOCK 
            MVI     A,2         CHANNEL 2 
            SCALL   .OPENR      OPEN FILE 
            JC      ERROR 
            . 
            . 
 
    MSGA    DB      12Q,'FILE NAME?',''+200Q 
    DEFALT  DB      'SY0TMP'    DEFAULT DEVICE AND EXTENSION 
    BUFFER  DS      20          FILE NAME BUFFER 
    FNAME   DS      'SY1:MYFILE.NEW',0   FILE NAME FOR CHANNEL 1 
    ....................................................................... 
 
 
    .OPENW - Open File for Write (Octal 43Q) 
    ======================================== 
 
    ***     .OPENW - OPEN FILE FOR WRITE 
    * 
    *       OPENW IS CALLED TO OPEN A CHANNEL FOR WRITE. 
    * 
    *       THE FILE IS ENTERED IN THE CHANNEL TABLE, BUT NOT ON THE 
    *       DISK.  IT WILL BE ENTERED IN THE DIRECTORY AT CLOSE TIME. 
    * 
    *       THE CALLER SUPPLIES A FILE NAME, A DEFAULT BLOCK FOR THE 
    *       DEVICE AND EXTENSION, AND A CHANNEL NUMBER. 
    * 
    *       DEFAULT BLOCK FORMAT: 
    * 

    .OPENW - Open File For Write (Octal 43Q)(Cont) 
    ============================================== 
 
    * 
    *       DB      'DDD'       DEFAULT DEVICE 
    *       DB      'XXX'       DEFAULT EXTENSION 
    * 
    *       ENTRY   (DE) = DEFAULT BLOCK ADDRESS 
    *               (HL) = NAME ADDRESS 
 
    *               (A) = CHANNEL NUMBER 
    *       EXIT    'C' CLEAR IF OK 
    *               (HL) = ADVANCED PAST FILE NAME 
    *               'C' SET IF ERROR 
    *                (A) = ERROR CODE 
    *       USES    ALL 
 
 
    Notes: 
    ------ 
    Use  the  .OPENW SCALL to open a file for writing.  When HDOS processes 
    the .OPENW SCALL, the file is opened with  a  "temporary"  name,  which 
    does  not  appear  in  the disk directory.  When the channel is closed, 
    HDOS will then enter the name in the directory.  If any  previous  file 
    by  that  name  existed, it will be automatically deleted at that time. 
    This procedure has three implications: 
     
         1.   You  cannot  modify  an  existing file by means of the .OPENW 
    SCALL!  .OPENW is intended for creating  new  files  or  replacing  old 
    ones. 
 
         2.   If  you  are replacing an existing file, there must be enough 
    free space to hold both the new version and the old one, as the old one 
    will  not  be  deleted  until the new one is closed.  You might want to 
    manually delete the old file first via .DELETE. 
     
         3.  If you do not properly close the channel, the new file will be 
    lost.  This is intended as a safety factor.  A previously existing file 
    by  that  name  will  not  be  destroyed  until  the  new  one has been 
    successfully completed.  If you should start to write a  file  by  some 
    name,  then  realize  that you already have a useful file by that name, 
    you can CTRL-Z out and still retain the old file. 
     
    HDOS  will not allow any one file to be open for both read and write at 
    the same time, nor may any one file be open for write on more than  one 
    channel.   If you attempt to do this, you will cause a "usage conflict" 
    error.  This means that you cannot open a file with  .OPENW  if  it  is 
    already open for write or update, or if it is open for read.           

    .OPENW - Open File For Write (Octal 43Q)(Cont) 
    ============================================== 
 
    The  examples  shown  above for .READ are applicable to .WRITE as well. 
    The following example illustrates opening a file on a non-disk  device, 
    "AT:."   Note that exactly the same procedure is followed.  In fact, in 
    the above example where the user types in a file name, he may  just  as 
    well have typed in "TT:" or "AT:" for a device specification.            
     
     
    **      EXAMPLE: 
 
            . 
 
            . 
            MVI     A,3         OPEN ON CHANNEL 3 
            LXI     D,DEFALT    POINT TO DEFAULT BLOCK 
            LXI     H,FNAME 
            SCALL   .OPENW 
            JC      ERROR       ERROR 
            . 
            . 
            . 
    DEFALT  DB      'SY0',0,0,0 UNUSED, BUT REQUIRED 
    FNAME   DB      'AT:',0     NAME AND EXTENSION MEANINGLESS 
    ....................................................................... 
 
 
    .OPENU - Open File For Update (Octal 44Q) 
    ========================================= 
 
    ***     .OPENU - OPEN FILE FOR UPDATE 
    * 
    *       OPENU IS CALLED TO OPEN A CHANNEL FOR UPDATE. 
    * 
    *       UPDATE IS JUST LIKE READ, BUT THE FILE MAY BE WRITTEN 
    *       ALSO. 
    * 
    *       THE CALLER SUPPLIES A FILE NAME.  A DEFAULT BLOCK FOR THE 
    *       DEVICE AND EXTENSION, AND A CHANNEL NUMBER. 
    * 
    *       DEFAULT BLOCK FORMAT: 
    * 
    *       DB      'DDD'       DEFAULT DEVICE 
    *       DB      'XXX'       DEFAULT EXTENSION 
    * 
    *       ENTRY   (DE) = DEFAULT BLOCK ADDRESS 
    *               (HL) = NAME ADDRESS 
    *               (A) = CHANNEL NUMBER 
    * 

    .OPENU - Open File For Update (Octal 44Q)(Cont) 
    =============================================== 
 
    * 
    *       EXIT    'C' CLEAR IF OK 
    *               (HL) = ADVANCED PAST FILE NAME 
    *               'C' SET IF ERROR 
    *                (A) = ERROR CODE 
    *       USES    ALL 
 
 
    Notes: 
    ------ 
    Use  .OPENU  to  open  a file for update.  This means that a previously 
    existing disk file is opened for both read and write.  When opened, the 
    file is positioned at sector zero.            
     
    If  the  channel  is positioned over an existing sector and you issue a 
    .WRITE,  then that  sector  will  be  rewritten.   If  the  channel  is 
    positioned  at the end of the file, the file will be extended.  You can 
    use the .POSIT SCALL to position the channel at the end  of  the  file. 
    Thus,  the  .OPENU  and  .POSIT  SCALL combination allows you to append 
    information onto an existing file.             
 
     
                                     NOTE 
              Always  close  a file  that  you opened for update. 
              Failure to do so causes undefined results.  Failing 
              to  close  the  channel  properly  can  also  cause 
              "orphaned" sectors, which are not being used  by  a 
              file,  nor  are  they  in the free list.  HDOS will 
              automatically recover these orphans when  the  disk 
              is  next  mounted  or booted and return them to the 
              free list.             
 
    The  examples  used  for .OPENR on page 13-31 also apply to .OPENU.  Of 
    course, there are some differences. 
     
         1.   The file opened must already exist.             
 
         2.   The  file  must  reside on a mass storage device which can be 
    both read and written (i.e., not write-protected). 
    ....................................................................... 

    .OPENC - Open Contiguous File for Write (Octal 45Q) 
    ==================================================== 
 
    ***     .OPENC - OPEN CONTIGUOUS FILE FOR WRITE 
    * 
    *       OPENC IS CALLED TO OPEN A CHANNEL FOR WRITE. 
    * 
    *       THE FILE IS ENTERED IN THE CHANNEL TABLE, BUT NOT ON THE 
    *       DISK.  IT WILL BE ENTERED IN THE DIRECTORY AT CLOSE TIME. 
    * 
    *       THE CALLER SUPPLIES A FILE NAME, A DEFAULT BLOCK FOR THE  
    *       DEVICE AND EXTENSION, AND A CHANNEL NUMBER. 
    * 
    *       DEFAULT BLOCK FORMAT: 
    * 
    *       DB      'DDD'           DEFAULT DEVICE 
    *       DB      'XXX'           DEFAULT EXTENSION 
    * 
    *       ENTRY   BC   = SECTOR COUNT 
    *               (DE) = DEFAULT BLOCK ADDRESS 
    *               (HL) = NAME ADDRESS 
    *               A    = CHANNEL NUMBER 
    *       EXIT    'C' CLEAR IF OK 
    *               (HL) = ADVANCED PAST FILE NAME 
    *               'C' SET IF ERROR 
    *               (A)  = ERROR CODE 
    *       USES    ALL 
 
 
    Notes: 
    ------ 
    The  .OPENC  SCALL  is used in a manner similar to the .OPENW SCALL, in 
    that it is used to open a file for output.  The difference is that HDOS 
    will  insure  that  the  file  is  written  with  all  disk  sectors in 
    contiguous order.  If there are not enough contiguous sectors available 
    on  the diskette (due to fragmentation from prior creation and deletion 
    of files), the .OPENC SCALL will fail.  Naturally, HDOS  will  need  to 
    know  in  advance  exactly  how  big  the  file will be, so that it can 
    determine if a large enough contiguous area  exists  on  the  diskette. 
    The  file  size  (in sectors) is passed to HDOS in register pair BC for 
    the .OPENC SCALL, whereas the .OPENW SCALL does not care what value  is 
    present in register pair BC when it is called.  This is one of the very 
    useful undocumented SCALLs.  HDOS primarily uses it for the creation of 
    the  operating  system  files during SYSGEN, but you can use it anytime 
    you want to insure that the files you create occupy contiguous sectors, 
    as  long  as  you  accept the restrictions that this imposes -- (1) you 
    need to know the file size in advance, and (2) you may fail to open the 
    file,  even  if  there  is room on the diskette, if there is not enough 
    CONTIGUOUS space.  
    ....................................................................... 

    .CLOSE - Close Channel (Octal 46Q) 
    ================================== 
 
    ***     .CLOSE - PROCESS CLOSE SCALL. 
    *   
    *       CLOSE PROCESSING DEPENDS UPON THE FILE AND DEVICE TYPE. 
    * 
    *       FOR A WRITE/DIRECTORY TYPE, THE DIRECTORY IS SEARCHED 
    *       FOR A PREVIOUS ENTRY.  IF FOUND, IT IS DELETED.  THE NEW 
    *       ENTRY IS THEN INSERTED. 
    * 
    *       FOR A UPDATE/DIRECTORY TYPE, THE PREVIOUS ENTRY IS UPDATED. 
    * 
    *       FOR ALL FILES, THE DRIVER IS CALLED WITH THE DC.CLO 
    *       FUNCTION.  THE CHANNEL IS RELEASED. 
    * 
    *       ENTRY   (A) = CHANNEL # 
    *       EXIT    'C' CLEAR IF OK 
    *               'C' SET IF ERROR 
    *                (A) = CODE 
    *       USES    ALL 
 
 
    Notes: 
    ------ 
    Use  the  .CLOSE  SCALL  to  close a channel when you are done with it. 
    Always close all channels your program has opened, with two exceptions: 
     
         1.   HDOS enters your program with channel -1 open on your program 
    load file.  If you do not use this channel, you need not  close  it  -- 
    HDOS will perform the close on it automatically. 
     
         2.   Scratch  files  which  were  created via .OPENW, which are no 
    longer needed, need not be closed.  See ".CLEAR," on page 13-49. 
 
 
    **      EXAMPLE: 
 
            . 
            . 
            MVI     A,1 
            SCALL   .CLOSE      CLOSE CHANNEL 1 
            JC      ERROR 
            MVI     A,2 
            SCALL   .CLOSE      CLOSE CHANNEL 2 
            JC      ERROR       IF ERROR 
    ....................................................................... 

    .POSIT - Position Disk File (Octal 47Q) 
    ======================================= 
 
    ***     .POSIT - POSITION FILE. 
    * 
    *       LXI     B,POSITION 
    *       MVI     A,CHANNEL NUMBER 
    *       SCALL   .POSIT 
    * 
    *       ENTRY   (A) = CHANNEL NUMBER 
    *               (BC) = SECTOR NUMBER TO POSITION BEFORE 
    *       EXIT    'C' CLEAR IF OK 
    *               'C' SET IF ERROR 
    *                (A) = ERROR CODE 
    *                (A) = EC.EOF IF OFF END 
    *                 (BC) = SECTORS UNSKIPPED (REMAINDER OF COUNT) 
    *                 FILE POSITIONED AT EOF 
    *       USES    ALL 
 
 
    Notes: 
    ------ 
    Use the .POSIT SCALL to position the "channel cursor."  Since each read 
    or write on a file, via a channel, must transfer in sector  (or  multi- 
    sector)  lots, the channel's current position in the file is simply the 
    logical  sector  number next to be read or written.  This sector number 
    has no relation to actual physical sector numbers.  The first sector in 
    a  file  is  sector zero, the next is sector 1, the last sector in an n 
    sector file is n-1. 
     
                                     NOTE 
              The  .POSIT  SCALL  positions  the  channel  (file) 
              before the specified sector.  Thus, a .POSIT  to  0 
              positions  the  channel  before sector 0, so that a 
              one-sector read will return sector 0.  To  position 
              the  channel  at  the  end  of a file, .POSIT to n, 
              where n is the number of sectors in the  file.   If 
              you  do  not  know  how long the file is, .POSIT to 
              65535 (377377A), verify that an  EC.EOF  error  was 
              flagged,  and  then compute the file size as SIZE = 
              65535-(BC).             
               
 
    Thus,  when a file is first opened via .OPENR, .OPENW, or .OPENU, it is 
    positioned at sector 0.  The first read or write of m sectors will read 
    or  write  sectors  0 through m-1.  This is a normal sequential access. 
    For example, when reading, each one-sector read will  return  the  next 
    sector in the file. 

    .POSIT - Position Disk File (Octal 47Q) 
    ======================================= 
 
    You can use the .POSIT SCALL to set this "sector cursor" at any spot in 
    the file.  Positioning  a  file  at  sector  0  is  the  equivalent  of 
    rewinding  it.   A  file  may  be positioned at its end, so a read will 
    return end-of-file, and a write will extend the file.  It  may  not  be 
    positioned  after  the  last  sector +1  in  an  attempt  to extend the 
    file.  Files may be extended only via .WRITE SCALLs.            
     
    The  .POSIT SCALL strengthens the similarity between .OPENW and .OPENU. 
    If you have opened a file via .OPENW, you may use  .POSIT  to  position 
    the  channel  cursor to allow you to rewrite any sector in the file, at 
    any time.  If you then wish to add some more sectors to  the  end,  you 
    can  position  to the end of the file, and .WRITE some more.  Also note 
    that you can change the value of any byte or bytes in a file  open  for 
    write  or  update  by positioning before the proper sector, reading the 
    sector, modifying it, repositioning it  over  again,  and  writing  the 
    sector back.           
     
     
    **      EXAMPLE 1: REWINDING A FILE AFTER READING IT 
      
            <OPEN AND READ A FILE ON CHANNEL 1> 
 
            . 
            . 
            . 
            LXI     B,0         BEFORE SECTOR 0 
            MVI     A,1         CHANNEL 1 
            SCALL   .POSIT      POSITION 
            JC      ERROR 
 
            <READ THE FILE OVER AGAIN> 
 
    **      EXAMPLE 2: REPLACING A SECTOR IN A FILE BEING WRITTEN 
 
            <OPEN THE FILE VIA .OPENW > 
            MVI     A,2         CHANNEL 2 
            LXI     B,256*10    WRITE 10 SECTORS 
            LXI     D,BUFFER    FROM BUFFER 
            SCALL   .WRITE 
            JC      ERROR 
            LXI     B,1         PREPARE TO RE-WRITE 2ND SECTOR IN FILE 

    .POSIT - Position Disk File (Octal 47Q)(Cont) 
    ============================================= 
 
            MVI     A,2 
            SCALL   .POSIT 
            JC      ERROR 
            MVI     A,2         CHANNEL 2 
            LXI     B,256 
            LXI     D,BUFFER2 
            SCALL   .WRITE      WRITE DIFFERENT DATA 
            MVI     A,2         CHANNEL 2 
            LXI     B,-1        POSITION AT END OF FILE 
            SCALL   .POSIT      WILL RETURN EOF ERROR 
            CPI     EC.EOF 
            JNE     ERROR       OTHER ERROR 
 
            < FURTHER WRITES WILL APPEND TO END OF FILE > 
 
    **      EXAMPLE 3: INCREMENTING BYTE 7423 IN FILE "DATA.RAW" 
 
            MVI     A,0         OPEN ON CHANNEL 0 
            LXI     D,DEFALT     
            LXI     H,FNAME 
            SCALL   .OPENU      OPEN FOR UPDATE 
            JC      ERROR 
 
    *       POSITION FOR READ 
 
            LXI     H,7423      (H) = SECTOR NUMBER 
                                (L) = BYTE INDEX 
            MOV     C,H 
            MVI     B,0         (BC) = SECTOR NUMBER 
            PUSH    H           SAVE (HL) 
            MVI     A,0 
            SCALL   .POSIT      POSITION 
            JC      ERROR 
 
 
    *       READ SECTOR INTO WORK AREA 
 
            MVI     A,0         (A) = CHANNEL 
            LXI     B,256 
            LXI     D,BUFFER 
            SCALL   .READ 
            JC      ERROR 

    .POSIT - Position Disk File (Octal 47Q)(Cont) 
    ============================================= 
 
    *       INCREMENT BYTE 
 
            POP     B           (B) = SECTOR, (C) = BYTE INDEX 
            LXI     H,BUFFER 
            MOV     A,C         (A) = BYTE INDEX 
            CALL    $DADA       ADD (A) INTO (HL) (ROUTINE IN 
                                H17 ROM) 
            INR     M           INCREMENT BYTE IN BUFFER 
 
    *       POSITION FOR RE-WRITE 
 
            MOV     C,B 
            MVI     B,0         (BC) = SECTOR NUMBER 
            MVI     A,0 
            SCALL   .POSIT 
            JC      ERROR 
 
    *       WRITE BACK OUT 
 
            MVI     A,0         (A) = CHANNEL 
            LXI     B,256 
            LXI     D,BUFFER 
            SCALL   .WRITE 
            JC      ERROR 
 
    *       CLOSE FILE 
 
            MVI     A,0 
            SCALL   .CLOSE 
            JC      ERROR 
            . 
            . 
            . 
    DEFALT  DB      'SY0',0,0,0 
    FNAME   DB      'SY0:DATA.RAW',0 
    BUFFER  DS      256 
 
    ....................................................................... 

    .DELETE - Delete Disk File (Octal 50Q) 
    ====================================== 
 
    **      DELETE - PROCESS DELETE COMMAND. 
    *                            
    *       ENTRY   (HL) = NAME STRING 
    *               (DE) = DEFAULT BLOCK 
    *       EXIT    'C' CLEAR IF OK 
    *               'C' SET IF ERROR 
    *                (A) = CODE 
    *       USES    ALL 
 
 
    Notes: 
    ------ 
    Use the .DELETE SCALL to delete a disk file.  The format of the call is 
    similar to that of .OPENW, except that no channel number is  specified. 
    Note that deleting a file is considered to be a form of writing, so the 
    file must not be open on any channel for reading or  writing,  as  that 
    would cause a "file usage conflict."             
 
 
    **      EXAMPLE: 
 
            DELETE FILE "SY0:TEMP.TMP" 
 
    DELTEMP LXI     H,NAME 
            LXI     D,DEFALT 
            SCALL   .DELETE 
            JC      ERROR 
            . 
            . 
    NAME    DB      'TEMP.TMP',0     FILE NAME 
    DEFALT  DB      'SY0XXX'         DEFAULT DEVICE, DEFAULT EXTENSION 
    ....................................................................... 
 
    .RENAME - Rename A File (Octal 51Q) 
    =================================== 
 
    ***     .RENAME - PROCESS RENAME FUNCTION. 
    * 
    *       RENAME RENAMES A FILE ON A DIRECTORY DEVICE. 
    * 
    *       * NOTE * RENAME DOES NOT CHECK TO SEE IF THE NEW NAME 
    *       ALREADY EXISTS--THIS IS THE RESPONSIBILITY OF THE CALLER! 
    * 
    *       ENTRY   (HL) = NAME STRING 
    *               (DE) = DEFAULT BLOCK 
    *               (BE) = NEW NAME STRING 
    * 

    .RENAME - Rename A File (Octal 51Q)(Cont) 
    ========================================= 
 
    * 
    *       EXIT    'C' CLEAR IF OK 
    *               'C' SET IF ERROR                                   
    *                (A) = CODE 
    *       USES    ALL 
 
 
    Notes: 
    ------ 
    Use the .RENAME SCALL to change the name of a file on disk.  A renaming 
    is considered a form of writing on a file, so the same "usage conflict" 
    restrictions  apply: the file to be renamed must not be open on another 
    channel.  Two other restrictions exist: 
     
         1.   A  file  with  the  "new name" must not already exist on that 
    device.  .RENAME, unfortunately, does not check this for  you,  so  you 
    must  check  yourself  by  trying  to .OPENR  the file before doing the 
    .RENAME.  Also, .RENAME will allow you to designate two files on a disk 
    with the same name.  The results of this will be disasterous. 
     
         2.   The "name string" and the "new name string" must both specify 
    the  same  device  (SY0:,   SY1:,   SY2:,   DK0:,   DK1:,   or   DK2:). 
    Alternatively,  both files may use the default device, which may be any 
    valid HDOS disk drive name. 
 
     
                                     NOTE 
              The default block device and  extension apply  only 
              to the old file name, not to the new name.  The new 
              file   name  must  be  fully  specified,  including 
              device, file name, and extension, if there is to be 
              one.             
 
 
    **      EXAMPLE: 
 
    *       RENAME 'SY1:SORT.ASM' TO 'SY1:SORT.BAK' 
 
            LXI     B,NEWNAM 
            LXI     D,DEFALT 
            LXI     H,OLDNAM 
            SCALL   .RENAME 
            JC      ERROR 
            . 
            . 
    NEWNAM  DB      'SY1:SORT.BAK',0     NO DEFAULTS ALLOWED 
    OLDNAM  DB      'SORT',0             USE DEFAULT DEVICE AND EXTENSION 
    DEFALT  DB      'SY1ASM',0           DEFAULT DEVICE AND EXTENSION 
    ....................................................................... 

    .SETTOP - Set Top of User Memory (Octal 52Q) 
    ============================================ 
 
    ***     .SETTOP - SET TOP OF USER MEMORY 
    * 
    *       SETTOP IS CALLED TO NOTIFY THE SYSTEM OF A NEW 
    *       MEMORY LIMIT ADDRESS.  IF NECESSARY, THE OVERLAYS 
    *       WILL BE UNLOADED. 
    * 
    *       ENTRY   (HL) = NEW ADDRESS 
    *       EXIT    (PSW) = 'C' CLEAR IF OK 
    *                       'C' SET IF TOO HIGH 
    *                        (A) = ERROR CODE 
    *                        (HL) = MAXIMUM ADDRESS 
    *       USES    ALL 
 
 
    Notes: 
    ------ 
    Use  the  .SETTOP  SCALL to set the top of the user memory area.  Since 
    HDOS sets the top of memory to the last address in your  program,  most 
    programs  do not need to use .SETTOP.  Programs which need large buffer 
    areas should not declare them with DS statements, since  the  generated 
    binary file will be excessively large.  Instead, they should define the 
    areas via EQU statements, and use the  .SETTOP  SCALL  to  request  the 
    needed space from HDOS.            
 
    Note that by requesting the impossible (65535 bytes), you can determine 
    the actual  maximum  memory  available  from  the  error  return.   The 
    following  information  regarding  overlays  pertains  to  HDOS 2.0 and 
    below.  [If you want to request maximum memory, but avoid swapping  the 
    overlays,  the  approved  method  is  to  first load both overlays (see 
    .LOADO for details) and then make the memory request.] 
     
    **      EXAMPLE 1: GETTING MAXIMUM MEMORY WITHOUT SWAPPING 
 
            MVI     A,OVL0      LOAD OVERLAY 0 
            SCALL   .LOADO 
            JC      ERROR 
    * 
            MVI     A,OVL1      LOAD OVERLAY 1 
            SCALL   .LOADO 
            JC      ERROR 
    * 
            LXI     H, -1       CAUSE DELIBERATE ERROR 
            SCALL   .SETTOP     .. TO GET MAX IN (HL) 
            LXI     D,-10       SUBTRACT 'SLOP' FACTOR 
            DAD     D 
            SHLD    MAXMEM      SAVE MAX MEMORY 
            SCALL   .SETTOP     NOW ASK FOR THE MAX ALLOWABLE 
            JC      ERROR       SHOULD NOT HAPPEN 

    .SETTOP - Set Top Of User Memory (Octal 52Q)(Cont) 
    ================================================== 
            . 
            . 
    MAXMEM  DS      2           MEMORY LIMIT 
 
                                                   
    **      EXAMPLE 2: GETTING ABSOLUTE MAXIMUM MEMORY 
    *                  ( ENTER HERE WITHOUT LOADING OVERLAYS ) 
 
            LXI     H,-1        IMPOSSIBLE AMOUNT 
            SCALL   .SETTOP     WILL FAIL ... 
            SHLD    MAXMEM      SAVE RESULT 
            SCALL   .SETTOP     ASK FOR MAX 
            JC      ERROR       SHOULD NOT HAPPEN 
            . 
            . 
    MAXMEM  DS      2           MEMORY LIMIT 
 
    *       REMEMBER IF THE .SETTOP IS SUCCESSFUL, 
    *       THE CONTENTS OF (HL) ARE MEANINGLESS. 
    ....................................................................... 
 
    .DECODE - Decode File Name (Octal 53Q) 
    ====================================== 
 
    ***     .DECODE - PROCESS DECODE SCALL. 
    * 
    *       DECODE DECODES THE SUPPLIED FILE NAME 
    *       INTO A BLOCK IN THE FORM: 
    * 
    *       DS      1           RESERVED 
    *       DS      2           DEVICE NAME 
    *       DS      1           DEVICE UNIT 
    *       DS      8           FILE NAME 
    *       DS      3           FILE EXTENSION 
    *       DS      4           RESERVED 
    * 
    *       ENTRY   (BC) = AREA FOR TABLE TO BE WRITTEN 
    *               (DE) = DEFAULT LIST 
    *               (HL) = NAME ADDRESS  
    *       EXIT    'C' CLEAR IF OK 
    *               'C' SET IF ERROR 
    *                (A) = ERROR CODE 
    *       USES    ALL 

    .DECODE - Decode File Name (Octal 53Q)(Cont) 
    ============================================ 
 
    Notes: 
    ------ 
    Use  the  .DECODE  SCALL  to  decode  an  ASCII  file descriptor into a 
    formatted block.  The fields in the block  contain  the  device,  unit, 
    name,  and extension values from the file descriptor.  The fields are 0 
    filled.  This function is useful for programs which wish to in some way 
    examine the file name, extension, or device specification without going 
    to the work of manually cracking the file descriptor.  For example,  if 
    your  program  reads  a file descriptor from the console, then wants to 
    know if the extension is "ABC," it might use the .DECODE SCALL to crack 
    out the extension field.             
     
    **      EXAMPLE: SEE IF USER TYPED DEVICE CODE 'TT:' 
 
            < READ LINE FROM CONSOLE INTO *LINE* > 
 
            LXI     B,BUFFER 
            LXI     D,DEFALT 
            LXI     H,LINE 
            SCALL   .DECODE     DECODE SUPPLIED FILE NAME 
            JC      ERROR       ILLEGAL NAME 
            LXI     B,3         COMPARE 3 BYTES 
            LXI     D,BUFFER+1  (DE) = SUPPLIED DEVICE NAME 
            LXI     H,TTSTR 
            CALL    #COMP       COMPARE STRINGS (ROUTINE 
                                IN H17 ROM) 
            JNE     NOTTT       FILE NOT ON TT: 
            JMP     GOTTT       NAME DID SPECIFY TT: 
 
    BUFFER  DS      19          ROOM FOR REPLY DATA 
    LINE    DS      80          USER SUPPLIED FILE NAME 
    TTSTR   DB      'TT',0      NAME AND UNIT IF DEVICE WAS 'TT:' 
    ....................................................................... 

    .NAME - Get File Name from Channel (Octal 54Q) 
    ============================================== 
 
    ***     .NAME - PROCESS NAME SCALL. 
    * 
    *        THE NAME SCALL RETURNS THE DEVICE, FILE NAME, AND 
    *        FILE EXTENSION OF AN OPEN CHANNEL. 
    * 
    *        THE INFORMATION IS OBTAINED FROM THE CHANNEL TABLE. 
    *        WHICH WAS SET UP UPON FILE OPEN. 
    * 
    *        ENTRY  (A) = CHANNEL NUMBER 
    *               (DE) = ADDRESS FOR DEVICE AND EXTENSION 
    *                      (DEFAULT BLOCK FORMAT) 
    *               (HL) = ADDRESS FOR NAME (8 CHARACTERS 
    *                      FOLLOWED BY 00 BYTE) 
    *       EXIT    'C' CLEAR IF OK 
    *               'C' SET IF ERROR 
    *                (A) = ERROR CODE 
    *       USES    ALL 
 
 
    Notes: 
    ------ 
    Use  the .NAME SCALL to recall the name which was supplied to HDOS when 
    the channel was opened.  This is mainly used when an error  message  is 
    prepared after HDOS has flagged an error on a channel operation.        
     
    **      EXAMPLE: ERROR PRINTING PROGRAM. 
    * 
    *       THIS ROUTINE PRINTS AN ERROR MESSAGE FOR A  
    *       FILE OPERATION GONE WRONG. 
    * 
    *       ENTRY   (A) = ERROR NUMBER 
    *               (CURCHAN) = CHANNEL NUMBER USED IN 
    *                           FAILED OPERATION 
    *       EXIT    ... 
 
    ERROR   PUSH    PSW         SAVE ERROR CODE 
            LXI     H,ERRORA 
            SCALL   .PRINT      PRINT 'ERROR - ' 
            POP     PSW         (A) = CODE 
            MVI     H,07Q       BELL AFTER ERROR CODE 
            SCALL   .ERROR      PRINT ERROR 
            LXI     H,ERRORB 
            SCALL   .PRINT      PRINT ' ON FILE ' 
            LDA     CURCHAN     (A) = CHANNEL NUMBER 
            LXI     D,ERRDFB    (DE) = ADDRESS FOR DEVICE AND EXTENSION 
 
            LXI     H,ERRNAM    (HL) = ADDRESS FOR NAME 
            SCALL   .NAME       GET FILE NAME 

    .NAME - Get File Name From Channel (Octal 54Q)(Cont) 
    ==================================================== 
 
    *       MANIPULATE DEVICE, NAME, AND EXTENSION INTO 
    *       PRESENTABLE FORMAT, AND PRINT ON CONSOLE 
 
            . 
            . 
            . 
    ERRORA  DB      D12Q,'ERROR -',' '+200Q 
    ERRORB  DB      ' ON FILE ',' '200Q 
    ERRDFB  DS      6               DEVICE AND EXTENSION FOR BAD FILE 
    ERRNAM  DS      9               NAME FOR BAD FILE 
    .......................................................................  

    .CLEAR - Clear I/O Channel (Octal 55Q) 
    ====================================== 
 
    ***     CLEAR - CLEAR I/O CHANNEL. 
    *        
    *       CLEAR IS CALLED TO CLEAR AN I/O CHANNEL.  IF THE 
    *       CHANNEL IS CLOSED, NO ACTION IS PERFORMED.  IF THE 
    *       CHANNEL IS OPEN, IT IS FLAGGED CLOSED.  THE RESULTS 
    *       OF THIS OPERATION DEPEND UPON THE TYPE OF FILE: 
    * 
    *       OPEN FOR                ACTION 
    * 
    *       READ                SAME AS .CLOSE 
    * 
    *       WRITE               FILE IS FORGOTTEN.  ANY WRITTEN 
    *                           DISK BLOCKS ARE RESTORED TO THE 
    *                           FREE POOL. 
    * 
    *       UPDATE              REPLACED SECTORS REMAIN REPLACED. 
    *                           APPENDED SECTORS ARE LOST UNTIL 
    *                           NEXT BOOT.  FILE STARY AT PREVIOUS 
    *                           LENGTH. 
    * 
    *       THE DEVICE DRIVER IS NOT INFORMED OF THE CLOSING. 
    * 
    *       SCALL   .CLEAR 
    * 
    *       ENTRY   (A) = CHANNEL NUMBER 
    *       EXIT    'C' CLEAR IS OK 
    *               'C' SET IF ERROR 
    *                (A) = ERROR CODE 
    *       USES    ALL 
 
 
    Notes: 
    ------ 
    Use  the  .CLEAR  SCALL  to  free up a channel without closing it.  The 
    actions discussed above merely document  the  current  results  of  the 
    .CLEAR  SCALL;  they may not stay the same for future releases of HDOS. 
    There is only one supported use of the .CLEAR SCALL, which is to delete 
    temp  files.   A temp work file is created by means of an .OPENW SCALL. 
    You  need not worry about name conflicts, as any  preexisting file with 
    the  same file name will not be disturbed by the .OPENW.  However, when 
    you are done, you do not want to .CLOSE then .DELETE  the  file,  since 
    this  would  destroy  any  preexisting file by that name.  In that case 
    use .CLEAR on the channel to free up the channel and release  the  used 
    disk sectors.  

    .CLEAR - Clear I/O Channel (Octal 55Q)(Cont) 
    ============================================ 
 
    **      EXAMPLE: CREATING, USING, AND DESTROYING A SCRATCH FILE 
 
            MVI     A,0         USE CHANNEL 0 
            LXI     D,DEFALT 
            LXI     H,SCRNAME 
            SCALL   .OPENW      OPEN TO SCRATCH FILE 
            JC      ERROR 
            . 
            . 
            < WRITE DATA ON SCRATCH FILE > 
            . 
            . 
            MVI     A,0 
            LXI     B,0 
            SCALL   .POSIT      REWIND SCRATCH FILE 
            . 
            . 
            < READ DATA FROM SCRATCH FILE > 
            . 
            . 
            MVI     A,0 
            SCALL   .CLEAR      DESTROY SCRATCH FILE 
            . 
            . 
    SCRNAME DB      'SY0:TEMP.TMP',0     ANY PRE-EXISTING TEMP.TMP 
                                         NOT AFFECTED 
    ....................................................................... 
 
 
 
    .CLEARA - Clear All Channels Except -1 (Octal 56Q) 
    ================================================== 
 
    ***     .CLEARA - CLEAR ALL CHANNELS. 
    * 
    *       CLEARA PERFORMS THE .CLEAR ACTION FOR ALL EXISTING CHANNELS, 
    *       EXCEPT CHANNEL 377Q, THE LOAD IMAGE CHANNEL. 
    * 
    *       ENTRY   NONE 
    *       EXIT    NONE 
    *       USES    ALL 

    .CLEARA - Clear All Channels Except -1 (Octal 56Q)(Cont) 
    ======================================================== 
 
    Notes: 
    ------ 
    The  .CLEARA  SCALL  is equivalent to calling the .CLEAR SCALL once for 
    every valid HDOS channel except for channel 377Q.  If you had  multiple 
    temporary  files  open simultaneously and wanted to scratch all of them 
    at once, this SCALL could be used  as  a  convenience.   Refer  to  the 
    section  on  .CLEAR  for  details  on  the actions taken by HDOS when a 
    channel is .CLEARed as opposed to being .CLOSEed.  There is  no  reason 
    to  avoid  the  use  of  this  SCALL, since it is fully functional.  It 
    appears to this author that the programmers working  on  HDOS  2.0  had 
    some plans to change the action of the .CLEAR SCALL in a future release 
    of HDOS, and they chose not to document the .CLEARA  SCALL  because  it 
    might also change.  Obviously, .CLEARA is only a 'convenience' for rare 
    situations when you have  multiple  temporary  output  files  open  (or 
    multiple files open for reading).  
 
    ....................................................................... 
 
 
    .ERROR - Print Error Message (Octal 57Q) 
    ======================================== 
 
    *       .ERROR - PRINT ERROR MESSAGE. 
    * 
    *       ERROR IS CALLED TO PRINT AN ERROR MESSAGE. 
    * 
    *       THE HDOS SYSTEM RETURNS ERROR CODE NUMBERS WHEN 
    *       IT DETECTS AN ERROR.  THE ERROR FUNCTION MAY BE 
    *       USED TO TYPE AN ALPHABETICAL EXPLANATION OF THE ERROR. 
    * 
    *       THE ERRORS ARE STORED IN THE FILE 'ERRORMSG.SYS' 
    *       ON THE SYSTEM DISK, ONE MESSAGE PER LINE. 
    *       THE LINES LOOK LIKE: 
    * 
    *       NNNTEXT 
    * 
    *       FOR EXAMPLE: 
    * 
    *       002END OF MEDIA 
    * 
    *       IF THE ERROR MESSAGE FILE CANNOT BE READ, OR THE  
    *       MESSAGE DOES NOT APPEAR, THE ERROR IS TYPED AS 
    * 
    *       'SYSTEM ERROR # NNN' 
    * 

    .ERROR - Print Error Message (Octal 57Q)(Cont) 
    ============================================== 
 
    * 
    *       ENTRY   (A) = ERROR CODE 
    *               (H) = TRAILING CHARACTER (TYPED AFTER MESSAGE) 
    *       EXIT    NONE 
    *       USES    ALL 
 
 
    Notes: 
    ------ 
    Use  the  .ERROR  SCALL to look up an error message in the system error 
    message file "ERRORMSG.SYS."  Since HDOS returns all error messages  as 
    numbers, this function allows you to easily inform the user in English, 
    and tell him just what went wrong.   Also  note  that  if  you  have  a 
    program which needs to generate a large number of messages, you can add 
    them to "ERRORMSG.SYS."  Of course, this is not a supported use of  the 
    .ERROR SCALL and may not work with future HDOS releases.         
     
    An  example  of  the use of the .ERROR SCALL is shown in the example of 
    the .NAME SCALL.           
    ....................................................................... 
 
 
    .CHFLG - Change File Flag(s) (Octal 60Q) 
    ======================================== 
 
    ***     .CHFLG - CHANGE FILE FLAGS. 
    * 
    *       CHFLG IS CALLED TO CHANGE THE FILE DESCRIPTION FLAGS 
    *       FOR A MASS STORAGE FILE.  ONLY CERTAIN FLAGS MAY BE 
    *       CHANGED: 
    * 
    *       FLAG    BIT         MEANING 
    * 
    *       DIF.SYS 200Q        IS SYSTEM FILE 
    *       DIF.LOC 100Q        LOCKED FOR CHANGE (SETABLE ONLY) 
    *       DIF.WP  040Q        IS WRITE PROTECTED 
    * 
    *       CHFLG WILL REFUSE THE OPERATION IF THE DIF.LOC BIT IS SET. 
    *        
    *       ENTRY   (B) = NEW BIT VALUES 
    *               (C) = CHANGE MASK (BIT SET FOR EVERY BIT 
    *                     TO REPLACE FROM (B)) 
    *               (DE) = DEFAULT BLOCK ADDRESS 
    *               (HL) = FILE NAME 
 
    * 

    .CHFLG - Change File Flag(s) (Octal 60Q)(Cont) 
    ============================================== 
 
    * 
    *       EXIT    'C' CLEAR, CHANGE DONE 
    *               'C' SET, ERROR 
    *                (A) = ERROR CODE 
    *       USES    ALL 
 
 
    Notes: 
    ------ 
    Use  the  .CHFLG  SCALL to change the attribute flags on a file.  These 
    flags are discussed in detail in  the  HEATH  HDOS  Software  Reference 
    Manual,  Chapter 4, SYSCMD/Plus, page 4-11.   The arguments are similar 
    to the .OPEN SCALLs.  Note that a two-byte "bits to  effect"  and  "new 
    bit  values"  scheme  is used, just as described earlier for the .CONSL 
    SCALL.             
     
                                     NOTE 
              You  can  use  the  .CHFLG SCALL to set the DIF.LOC 
              (LOCKed) flag on a file, but you cannot use  it  to 
              clear  the  flag.  Once the DIF.LOC flag is set, no 
              other flag  changes  may  be  made,  including  the 
              clearing  of  the DIF.LOC flag.  If the file is not 
              write-protected (DIF.WP not set),  you can copy  it 
              to  a temp file, delete the old LOCKed version, and 
              rename the temp file back.  If  the  file  is  both 
              LOCKed   and  write-protected,  then  it  is  there 
              "forever" or until the disk is reinitialized.      
 
 
    **      EXAMPLE: 
    * 
    *       WRITE PROTECT 'OUTPUT.DAT' 
      
    WRIPRO  MVI     B,DIF.WP    EFFECT WRITE PROTECT 
            MVI     C,DIF.WP    SET WRITE PROTECT 
            LXI     D,DEFALT 
            LXI     H,NAME 
            SCALL   .CHFLG 
            JC      ERROR       ERROR 
            . 
            . 
            . 
    NAME    DB      'SY1:OUTPUT.DAT',0     FILE NAME 
    ....................................................................... 

    .DISMT - Flag System Disk As Dismounted (61Q) 
    ============================================= 
 
    ***     .DISMT - FLAG SYSTEM DISK DISMOUNTED 
    * 
    *       THE DISMT FUNCTION IS USED WHEN THE SYSTEM DISK IS  
    *       ABOUT TO BE DISMOUNTED.  ANY HDOS FUNCTIONS WHICH 
    *       REQUIRE SYSTEM FILES WILL BE TREATED AS FATAL 
    *       SYSTEM ERRORS. 
    * 
    *       ENTRY   None                                       
    *       EXIT    S.SYSM = LWA OF FREE SPACE FOR USER 
    *               (HL) = (S.SYSM)             
    *       USES    ALL 
    ....................................................................... 
     
 
    .LOADD - Load Device Driver (Octal 62Q) 
    ======================================= 
 
    ***     .LOADD - LOAD DEVICE DRIVER 
    * 
    *       LOADD LOADS THE SPECIFIED DEVICE DRIVER. 
    * 
    *       ENTRY   (HL)    = DEVICE DRIVER DESCRIPTOR STRING. 
    *       EXIT    (PSW)   = 'C' CLEAR IF OK 
    *                         'C' SET IF ERROR 
    * 
    *       USES    ALL 
 
 
    Notes: 
    ------ 
    Use  the .LOADD system call to load a specified device driver in memory 
    without opening a file on the device.  Like  the  .LOADO  system  call, 
    this system call is not to be used when SY0: is to be dismounted.  If a 
    device driver is not in memory at the time SY0: is dismounted  (because 
    it  was  not  loaded  and  no channel is currently open on the device), 
    subsequent references  to  the  device  will  generate  unknown  device 
    errors.        
 
    Examples of this SCALL are found in Part 8, and in the example below: 
   
 
            LXI     H,DEVICE 
            SCALL   .LOADD 
            JC      ERROR 
            . 
            . 
    DEVICE  DB      'LP:',0 
    ....................................................................... 

                        PART 5 - RESIDENT SCALLS (Cont) 
                        +++++++++++++++++++++++++++++++ 
 
     
    .TASK - Communicates with TMG (Octal 101Q) 
    ========================================== 
 
    ***     TASK - EXTERNAL SCALL FOR THE TASK MANAGER. 
    * 
    *       TASK IS THE WAY THE USER COMMUNICATES WITH TMG, 
    *       THE TASK MANAGER. 
    * 
    *       ENTRY:  See TASKDEF.ACM 
    *       EXIT:   See TASKDEF.ACM 
    *       USES:   See TASKDEF.ACM 
 
    .TASK   EQU     101Q    41H     65 
    ....................................................................... 
 
 
    .TDU - External SCALL for Terminal DeBugging Utility (Octal 122Q) 
    ================================================================= 
 
    ***     TDU - EXTERNAL SCALL FOR TERMINAL DEBUGGING UTILITY. 
    * 
    *       TDU is a basic debugging task which allows the user to examine 
    *       and alter the state of his CPU. 
    * 
    *       THE USER MUST FIRST START THE TDU TASK WITH THE COMMAND: 
    * 
    *               ST TDU 
    * 
    *       ENTRY INTO TDU IS THEN ACHIEVED BY THE FOLLOWING: 
    * 
    *               . 
    *               user code 
    *               . 
    *               SCALL   .TDU            ; This is a breakpoint 
    *               . 
    *               user code 
    *               . 
    * 
    *       ENTRY:  NONE 
    *       EXIT:   NONE 
    *       USES:   Any registers the user alters 
 
    .TDU    EQU     122Q    52H     82 
    ....................................................................... 

    .LOG - External SCALL Used With the ECHO Task (177Q) 
    ==================================================== 
 
    ***     LOG - EXTERNAL SCALL USED WITH THE ECHO TASK. 
    * 
    *       LOG IS THE SOFTWARE METHOD TO TOGGLE THIS TASK ON AND OFF. 
    * 
    *       ENTRY:  A = 0   Turn the task off 
    *                 = 1   Turn the task on 
    *                 > 1   Will case error 
    *       EXIT:   PSW     'C' clear if NO error 
    *                       'C' set   if    error 
    *                         A = EC.ILC if ECHO is NOT resident 
    *                         A = -1     if illegal value was sent in A 
    *       USES:   A,F,H,L 
 
    .LOG    EQU     177Q    7FH     127 
    ....................................................................... 
 
 
    .MOUNT - Mount A Disk (Octal 200Q) 
    ================================== 
 
    ***     .MOUNT - MOUNT A DISK 
    * 
    *       MOUNT DISK ON SPECIFIED UNIT OF SELECTED DEVICE. 
    * 
    *       ENTRY   (HL)    = ADDRESS OF DEVICE SPECIFICATION 
    *       EXIT    (PSW)   = 'C' SET IF ERROR 
    *                         'C' CLEAR IF NO ERROR 
    *                          (A) = ERROR CODE 
    *                           'Z' CLEAR IF AN ABORT 
    *       USES    ALL 
 
 
    Notes: 
    ------ 
    Use  the  .MOUNT  system  call to mount additional devices.  The device 
    specified must not have a volume already mounted on it.  If it does,  a 
    successful  dismount  must  be issued before a .MOUNT may be processed. 
    The devices currently supported are SY0:, SY1:, SY2:, DK0:,  DK1:,  and 
    DK2:.  This system call also prints a message informing the user that a 
    volume has been mounted.  This message is the same one  used  with  the 
    HDOS  "MOUNT"  command.   This  call  will  also  verify  that the disk 
    structure is not corrupt.  If the disk  is  corrupt,  it  will  not  be 
    mounted, and an error message will be returned.  If you do not want the 
    message, you may issue the .MONMS system call.  
     
    For a more detailed example of .MOUNT, refer to Part 8. 
    ....................................................................... 

    .DMOUN - Dismount A Disk (Octal 201Q) 
    ===================================== 
 
    ***     .DMOUN - DISMOUNT A DISK 
    * 
    *       DISMOUNT DISK ON SELECTED DRIVE. 
    * 
    *       ENTRY   (HL)    = ADDRESS OF DEVICE SPECIFICATION  
    *       EXIT    (PSW)   = 'C' SET IF ERROR 
    *                         (A) = ERROR CODE 
 
    * 
    *       USES    ALL 
 
    Notes: 
    ------ 
    Use  the  .DEMOUN  system call to dismount diskettes.  After the volume 
    has  been  successfully  dismounted,  it  will  also  print  a  message 
    verifying  that  the  volume  has  been  dismounted.   The device to be 
    dismounted must have a volume  currently  mounted.   If  it  does  not, 
    .DMOUN returns an error.         
     
    If  the  volume to be dismounted is the system volume, you must observe 
    several precautions.  Device drivers not currently  in  memory  at  the 
    time  of  the  dismount  will  be  considered  nonexistent.  Subsequent 
    references to drivers so marked will generate  unknown  device  errors. 
    You  may  load a device driver by opening a channel on the device or by 
    ".LOADD"ing it.  Even if the current program will not use  the  device, 
    the  device must be loaded before you dismount the system volume if any 
    subsequent programs are to use it. 
     
    Before you dismount a disk, you must clear all of the I/O channels open 
    to that disk.  Remember that the program itself is left open on channel 
    -1 (377Q), and this channel must be  closed  before  you  dismount  the 
    system disk (SY0:).         
    ....................................................................... 
     
 
    .MONMS - Mount A Disk With No Message (Octal 202Q) 
    ================================================== 
 
    ***     MONMS - MOUNT/NO MESSAGE  
    * 
    *       MOUNT SPECIFIED UNIT OF SELECTED DEVICE WITHOUT ISSUING 
    *       A MOUNT MESSAGE. 
    * 
    *       ENTRY   (HL)    = ADDRESS OF DEVICE SPECIFICATION 
    *       EXIT    (PSW)   = 'C' SET IF ERROR 
    *                          (A) =ERROR CODE 
    *                         'C' CLEAR IF NO ERROR 
    *                          'Z' CLEAR IF AN ABORT 
    *       USES    ALL          

    .MONMS - Mount A Disk With No Message (Octal 202Q)(Cont) 
    ======================================================== 
 
    Notes: 
    ------ 
    In  versions  of  HDOS later than Version 1.5 the .MONMS system call is 
    identical to the .MOUNT system call, except that .MONMS prints no error 
    message.   In the future, this may not be the case.  In all likelihood, 
    this will be changed to  a  "quick"  mount  which  neither  prints  the 
    message  nor  verifies  the  disk  structure.   
    ....................................................................... 
 
 
    .DMNMS - Dismount A Disk With No Message (Octal 203Q) 
    ===================================================== 
 
    ***     .DMNMS - DISMOUNT A DISK WITH NO MESSAGE 
    * 
    *       DISMOUNT SELECTED UNIT OF SPECIFIED DEVICE WITHOUT 
    *       ISSUING MESSAGE. 
    * 
    *       ENTRY   (HL)    = ADDRESS OF DEVICE SPECIFICATION 
    *       EXIT    (PSW)   = 'C' SET IF ERROR 
    *                          (A)  = ERROR CODE 
    *       USES    ALL 
 
 
    Notes: 
    ------ 
    The  .DMNMS  system  call  is  virtually  identical to the .DMOUN call, 
    except for the printing of the dismount message.  In  future  releases, 
    this will probably be changed to some form of "quick" dismount.   
    ....................................................................... 
 
    .RESET - Reset A Disk (Octal 204Q) 
    ================================== 
 
    ***     RESET - RESET A DISK 
    * 
    *       RESET THE SPECIFIED UNIT OF THE SELECTED DEVICE 
    *       BY ISSUING A DISMOUNT FOLLOWED BY A MOUNT. 
    *       THE DEVICE NAME SHOULD BE IN THE SAME FORMAT AS 
    *       THAT EXPECTED BY A MOUNT AND DISMOUNT. 
    * 

    .RESET - Reset A Disk (Octal 204Q)(Cont) 
    ======================================== 
 
    *       ENTRY:  (HL) = ADDRESS OF DEVICE SPECIFICATION 
    *       EXIT:   (PSW) = 'C' CLEAR IF NO ERROR 
    *               'C' SET IF ERROR 
    *                     (A) = Error Code 
    *                  
    *       Uses:   ALL 
    ....................................................................... 
 
 
    .RESMNS - Reset A Disk With No Message (OCTAL 205Q) 
    =================================================== 
 
    ***     RESMNS - RESET A DISK WITH NO MESSAGE. 
    * 
    *       ENTRY:  (HL) = ADDRESS OF DEVICE SPECIFICATION 
    *               (BC) = ADDRESS OF USER MESSAGE 
    *                 0 = NO MESSAGE 
    *                -1 = USE STANDARD MESSAGE 
    *       EXIT:   'C' SET IF ERROR 
    *                (A) = ERROR CODE     
    *                                 
    *       Uses:   ALL 
    ....................................................................... 
 
 
    .DAD - Dismount all Disks (Octal 206Q) 
    ====================================== 
      
    ***     DAD     - DISMOUNT ALL DISKS 
    * 
    *       DAD DISMOUNTS ALL DISKS, THAT IS, ALL MOUNTED VOLUMES OF 
    *       ALL CURRENTLY MOUNTED DISKS.  THIS IS USUALLY USED IN 
    *       PREPARATION FOR DISMOUNTING THE SYSTEM DISK. 
    * 
    *       THIS IS INCLUDED AS A SYSTEM CALL SINCE THE .EXIT CALL 
    *       NEEDS THIS CODE.  EVERYONE ELSE MIGHT AS WELL HAVE ACCESS 
    *       TO IT. 
    * 
    *       ENTRY   NONE 
    *       EXIT    'NC' IF NO ERROR 
    *               'C' IF ERROR 
    *               A  = ERROR CODE 
    *       USES    ALL 
 
 
 
 
 
 



    CHAPTER THIRTEEN    HDOS PROGRAMMERS' REFERENCE MANUAL       PAGE 13-60 
    ================    ==================================       ========== 

    .DAD - Dismount All Disks (Octal 206Q)(Cont) 
    ============================================  
 
    Notes: 
    ------ 
    This SCALL is used to dismount ALL units of ALL directory devices known 
    to HDOS.  Note that this includes the system disk drive SY0: as well as 
    all  the  other disks that have been mounted.  Therefore, if this SCALL 
    is executed and the program issues an .EXIT SCALL afterwards,  it  will 
    require  the  user  to  reboot  HDOS.  Since HDOS 3.0 is already set to 
    stand-alone operation, reboot will not be necessary. 
 
    ----------------------------------------------------------------------- 
                       SUMMARY OF HDOS 3.0 SYSTEM CALLS 
    ----------------------------------------------------------------------- 
 
 Name   OCT HEX DEC Description 
      =======      ===   ===   ===    =========== 
 
 .EXIT  EQU  0Q  0H   0 Exit to SYSCMD.SYS 
 
 .SCIN  EQU  1Q  1H   1 System Console Input 
 
 .SCOUT EQU  2Q  2H   2 System Console Output 
 
 .PRINT EQU  3Q  3H   3 Print Text 
 
 .READ  EQU  4Q  4H   4 Read Disk Sector(s) 
 
 .WRITE EQU  5Q  5H   5 Write Disk Sector(s) 
 
 .CONSL EQU  6Q  6H   6 Console Attributes 
 
 .CLRCO EQU  7Q  7H   7 Clear Console Buffer 
 
 .LOADO EQU 10Q  8H   8 Load Overlay (OBSOLETE) 
 
 .VERS  EQU 11Q  9H   9 Version 
 
 .GDA  EQU 12Q  AH  10 Get Driver Address 
 
 .CRC16 EQU 13Q  BH  11 CRC Checksum of Block 
 
      .LINK    EQU   40Q   20H    32   Link to a Program 
 
 .CTLC  EQU 41Q 21H  33 Control A, B & C 
 
 .OPENR EQU 42Q 22H  34 Open for Read 
 
 .OPENW EQU 43Q 23H  35 Open for Write 

    ----------------------------------------------------------------------- 
                       SUMMARY OF HDOS 3.0 SYSTEM CALLS 
    ----------------------------------------------------------------------- 
 
 Name   OCT HEX DEC Description 
      =======          ===   ===   ===   =========== 
 
      .OPENU    EQU   44Q   24H    36   Open for Update 
 
 .OPENC  EQU 45Q 25H  37 Open Contiguous 
 
 .CLOSE  EQU 46Q 26H  38 Close a channel 
 
 .POSIT  EQU 47Q 27H  39 Position within a File 
 
 .DELETE   EQU 50Q 28H  40 Delete a File 
 
 .RENAME   EQU 51Q 29H  41 Rename a File 
 
 .SETTOP   EQU 52Q 2AH  42 Set Top of User Memory 
 
 .DECODE   EQU 53Q 2BH  43 Decode File Name 
 
 .NAME   EQU 54Q 2CH  44 Get Name from Channel 
 
 .CLEAR  EQU 55Q 2DH  45 Clear a Channel 
 
 .CLEARA   EQU 56Q 2EH  46 Clear All Channels except -1 
 
 .ERROR  EQU 57Q 2FH  47 Print Error Message 
 
 .CHFLG  EQU 60Q 30H  48 Change File Flag(s) 
 
 .DISMT  EQU 61Q 31H  49 Flag System Disk as Dismounted 
 
 .LOADD  EQU 62Q 32H  50 Load Device Driver 
 
      .TASK     EQU   101Q  41H    65   Used with Task Manager 
 
      .TDU              EQU   122Q  52H    82   Used with Terminal Debugger 
 
      .LOG      EQU   177Q  7FH   127   Used with Echo Task 
 
 .MOUNT  EQU 200Q 80H 128 Mount a Disk 
 
 .MONMS  EQU 202Q 82H 130 Mount a Disk - No Messages 
 
 .DMNMS  EQU 203Q 83H 131 Dismount a Disk - No Messages 
 
 .RESET  EQU 204Q 84H 132 Reset a Drive 
 
 .RESNMS   EQU 205Q 85H 133 Reset a Drive - No Messages 
 
 .DAD   EQU 206Q 86H 134 Dismount All Devices 
    *********************************************************************** 
 



    CHAPTER THIRTEEN    HDOS PROGRAMMERS' REFERENCE MANUAL       PAGE 13-62 
    ================    ==================================       ========== 
 
 
                       PART 6 - HDOS SYMBOL DEFINITIONS 
                       ++++++++++++++++++++++++++++++++ 
 
    As  we  stressed in earlier sections of this manual, there are numerous 
    advantages to using symbolic definitions when you  are  interfacing  to 
    the  operating system.  This section lists suggested common decks which 
    contain the appropriate symbolic definitions.              
     
    To obtain access to these definitions, simply insert the pseudo ops: 
 
        XTEXT    HDOSDEF                XTEXT    ASCII 
        XTEXT    HOSEQU                 XTEXT    ECDEF 
 
    into  the  initial  statements  of  your  program.  This will cause the 
    assembler to process, as required, the statements in the file HDOS.ACM, 
    thus defining those symbols for that assembly.              
     
    Note that the assembler will not normally list the contents of any file 
    read by XTEXT.  However, by using the 
 
        LON    C 
 
    pseudo op, or the  
 
        /LON:C 
 
    switch when you are using the assembler, you can cause a listing of all 
    files read by XTEXT to be written to the listing file.  
    ....................................................................... 
 
    The  following  data is extracted from selected source code .ACM files. 
    Refer  to  the  on-disk  library  of  source  code  files  for  further 
    information. 
                                    
             Data Source                                    Page 
             -----------                                    ---- 
    HDOS 3.0 COMMON DECK CONTENTS 
        HOSDEF.ACM Contents ............................... 13-63 
        HOSEQU.ACM Contents ............................... 13-64 
        CONSL SCALL Symbols ............................... 13-64 
        ASCII.ACM Contents ................................ 13-65 
        TYPTX.ACM Contents ................................ 13-66 
        MOVE.ACM Contents ................................. 13-67 
        ECDEF.ACM Contents ................................ 13-67 
 
    HDOS SYMBOL VALUES 
        ECDEF Symbol Definitions .......................... 13-69 
        HOSDEF Symbol Definitions ......................... 13-69 
        HOSEQU Symbol Definitions ......................... 13-70 

    HDOS 3.0 COMMON DECK CONTENTS 
    ============================= 
 
    HOSDEF.ACM Contents  
    ------------------- 
 
    HOSDEF  SPACE   3,10 
    **      HOSDEF - DEFINE HOS PARAMETER. 
    * 
 
    VERS    EQU     3*16+0      CURRENT VERSION = 3.0  
    SYSCALL EQU     377Q        SYSCALL INSTRUCTION 
            ORG     0 
 
    *       RESIDENT FUNCTIONS 
 
    .EXIT   DS      1       0Q  EXIT (MUST BE FIRST)  
    .SCIN   DS      1       1Q  SCIN            
    .SCOUT  DS      1       2Q  SCOUT             
    .PRINT  DS      1       3Q  PRINT            
    .READ   DS      1       4Q  READ            
    .WRITE  DS      1       5Q  WRITE            
    .CONSL  DS      1       6Q  SET CLEAR CONSOLE OPTIONS 
    .CLRCO  DS      1       7Q  CLEAR CONSOLE BUFFER 
    .LOADO  DS      1      10Q  LOAD AN OVERLAY (OBSOLETE)   /30a/ 
    .VERS   DS      1      11Q  RETURN HDOS VERSION NUMBER 
    .GDA    DS      1      12Q  GET DEVICE DRIVER ADDRESS    /30a/ 
    .CRC16  DS      1      13Q  CRC A BLOCK OF MEMORY        /30a/ 
    .SYSRES DS      1           PRECEDING FUNCTIONS ARE RESIDENT 
 
    *       HDOSOVL0.SYS FUNCTIONS (Note: HDOS 3.0 Has No Overlays) 
 
    .LINK   DS      1      40Q  LINK  
    .CTLC   DS      1      41Q  CTL-C 
    .OPENR  DS      1      42Q  OPENR 
    .OPENW  DS      1      43Q  OPENW          
    .OPENU  DS      1      44Q  OPENU  
    .OPENC  DS      1      45Q  OPENC        
    .CLOSE  DS      1      46Q  CLOSE  
    .POSIT  DS      1      47Q  POSITION  
    .DELET  DS      1      50Q  DELETE 
    .DELETE  EQU     .DELET                  
    .RENAM  DS      1      51Q  RENAME  
    .RENAME  EQU     .RENAM  
    .SETTP  DS      1      52Q  SETTOP 
    .SETTOP  EQU     .SETTP                      
    .DECODE DS      1      53Q  DECODE  
    .NAME   DS      1      54Q  NAME 
    .CLEAR  DS      1      55Q  CLEAR  
    .CLEARA DS      1      56Q  RESERVED  
    .ERROR  DS      1      57Q  LOOKUP ERROR  
 
    .CHFLG  DS      1      60Q  CHANGE FLAGS  
    .DISMT  DS      1      61Q  FLAG SYSTEM DISK DISMOUNTED 
    .LOADD  DS      1      62Q  LOAD DEVICE DRIVER 

    HDOS 3.0 COMMON DECK CONTENTS (Cont) 
    ==================================== 
 
    HOSDEF.ACM Contents (Cont) 
    -------------------------- 
 
    *      HDOSOVL1.SYS FUNCTIONS (NOTE: HDOS 3.0 Has No Overlays) 
 
                   ORG     200Q  
    .MOUNT         DS      1     200Q  MOUNT (MUST BE FIRST) 
    .DMOUN         DS      1     201Q  DISMOUNT  
    .DMOUNT        EQU     .DMOUN       
    .MONMS         DS      1     202Q  MOUNT/NO MESSAGE  
    .DMNMS         DS      1     203Q  DISMOUNT/NO MESSAGE  
    .RESET         DS      1     204Q  RESET = DISMOUNT/MOUNT UNIT  
    .RESNMS        DS      1     205Q  RESET/OPTIONAL MESSAGE 
    .DAD           DS      1     206Q  DISMOUNT ALL DISKS  
    ....................................................................... 
 
    HOSEQU.ACM Contents  
    ------------------- 
 
             SPACE   4,10 
    **      HDOS SYSTEM EQUIVALENCES.                             /3.0a/ 
    * 
 
            ORG     040100A 
 
    S.EXITA DS      8               ; Jump to System Exit 
    D.CON   DS      16              ; Disk Constants 
    SYDD    EQU     *               ; System Disk Entry Point 
    D.VEC   DS      24*3            ; H17 Disk Vectors 
    D.RAM   DS      31              ; H17 Disk Work Area 
    S.VAL   DS      36              ; SYSTEM VALUES 
    S.INT   DS      147             ; SYSTEM INTERNAL WORK AREAS 
    S.SOVR  DS      2               ; STACK OVERFLOW WARNING 
            DS      042200A-*       ; SYSTEM STACK 
    STACKL  EQU     *-S.SOVR        ; STACK SIZE 
    STACK   EQU     *               ; LWA+1 SYSTEM STACK 
    USERFWA EQU     *               ; USER FWA 
                                                                                 
    *       Ensure Compatibility 
 
            ERRNZ   040130A-SYDD 
            ERRNZ   040277A-S.VAL 
            ERRNZ   040343A-S.INT 
            ERRNZ   042200A-USERFWA 
    ....................................................................... 
 
    **      THE FOLLOWING SYMBOLS ARE USED BY THE .CONSL SCALL 
 
    CSL.ECH EQU     10000000B   SUPPRESS ECHO 
    CSL.RAW EQU     00000100B   RAW MODE I/O 
    CSL.WRP EQU     00000010B   WRAP LINES AT WIDTH 
    CSL.CHR EQU     00000001B   OPERATE IN CHARACTER MODE 

    HDOS 3.0 COMMON DECK CONTENTS (Cont) 
    ==================================== 
 
    **      THE FOLLOWING SYMBOLS ARE USED BY THE .CONSL SCALL (Cont) 
 
    I.CSLMD EQU     0           CONSOLE MODE 
 
    CTP.BKS EQU     10000000B   TERMINAL PROCESSES BACKSPACES 
    CTP.FF  EQU     01000000B   TERMINAL PROCESSES FORMFEED 
    CTP.MLI EQU     00100000B   MAP LOWER CASE TO UPPER ON INPUT 
    CTP.MLO EQU     00010000B   MAP LOWER CASE TO UPPER ON OUTPUT 
    CTP.2SB EQU     00001000B   TERMINAL NEEDS TWO STOP BITS 
    CTP.HHS EQU     00000100B   TERMINAL USES HARDWARE HANDSHAKE 
    CTP.BKM EQU     00000010B   MAP BKSP (UPON INPUT) TO RUBOUT 
    CTP.TAB EQU     00000001B   TERMINAL SUPPORTS TAB CHARACTERS 
 
    I.CONTY EQU     1           S.CONTY IS 2ND BYTE 
    I.CUSOR EQU     2           S.CUSOR IS 3RD BYTE 
    I.CONWI EQU     3           S.CONWI IS 4TH BYTE 
        
    CO.FLG  EQU     00000001B   CTL-O FLAG 
    CO.FLG  EQU     10000000B   CTL-S FLAG 
    ....................................................................... 
 
    ASCII.ACM Contents  
    ------------------ 
 
            SPACE   3,10 
    **      ASCII CHARACTER EQUIVALENCES. 
 
    NUL     EQU     000Q             ; null 
    BELL    EQU     007Q             ; bell 
    BKSP    EQU     010Q             ; backspace 
    BS      EQU     BKSP 
    TAB     EQU     011Q             ; horizontal tab 
    LF      EQU     012Q             ; line feed 
    NL      EQU     012Q             ; new line (HDOS) 
    FF      EQU     014Q             ; form feed 
    CR      EQU     015Q             ; carriage return 
    ESC     EQU     033Q             ; escape 
    DEL     EQU     177Q             ; delete 
 
    *       Specials 
 
    EOL     EQU     200Q             ; end of line flag 
    NULL    EQU     200Q             ; pad character 
    NUL2    EQU     0                ; ditto 
    RUBOUT  EQU     DEL              ; rubout/delete 
    C.SYN   EQU     026Q             ; SYNC 
    C.STX   EQU     002Q             ; STX 
    QUOTE   EQU     047Q             ; quote character (") 
    ENL     EQU     NL+EOL           ; NL + end-of-line flag 

    HDOS 3.0 COMMON DECK CONTENTS (Cont) 
    ==================================== 
 
    ASCII.ACM Contents (Cont) 
    ------------------------- 
 
    *       Control keys 
 
    CTLA    EQU     'A'-'@'          ; CTRL/A 
    CTLB    EQU     'B'-'@'          ; CTRL/B 
    CTLC    EQU     'C'-'@'          ; CTRL/C 
    CTLD    EQU     'D'-'@'          ; CTRL/D 
    CTLE    EQU     'E'-'@'          ; CTRL/E 
    CTLO    EQU     'O'-'@'          ; CTRL/O 
    CTLP    EQU     'P'-'@'          ; CTRL/P 
    CTLQ    EQU     'Q'-'@'          ; CTRL/Q 
    CTLR    EQU     'R'-'@'          ; CTRL/R 
    CTLS    EQU     'S'-'@'          ; CTRL/S 
    CTLX    EQU     'X'-'@'          ; CTRL/X 
    CTLZ    EQU     'Z'-'@'          ; CTRL/Z 
    ....................................................................... 
 
    TYPTX.ACM Contents  
    ------------------ 
 
    $TYPTX  SPACE   4,10 
    **      $TYPTX - TYPE TEXT. 
    * 
    *       $TYPTX IS CALLED TO TYPE A BLOCK OF TEXT TO THE SYSTEM 
    *       CONSOLE, WHERE THE BLOCK OF TEXT TO BE TYPED IMMEDIATELY 
    *       FOLLOWS THE SUBROUTINE CALL INSTRUCTION. 
    * 
    *       $TYPTX. PERFORMS THE SAME FUNCTION AS $TYPTX, EXCEPT THAT 
    *       USES REGISTER H,L AS A POINTER TO THE TEXT WHICH IS TO 
    *       BE TYPED. 
    * 
    *       IMBEDDED ZERO BYTES INDICATES A CARRIAGE RETURN LINE FEED. 
    *       A BYTE WITH THE 200Q BIT SET IS THE LAST BYTE IN THE MESSAGE. 
    * 
    *       EMTRY   (RET) = TEXT 
    *       EXIT    TO (RET+LENGTH) 
    *       USES    A,F 
                                        
    $TYPTX  EQU     031136A                 ; IN H17 ROM 
                    TAKES TEXT IMMEDIATELY 
 
    $TYPTX. EQU     031144A                 ; IN H17 ROM 
                    TAKES TEXT FROM HL REGISTER 
 
    EXAMPLE: 
 
    $TYPTX          TAB IN ONE TAB STOP, OR CALL TAB OVER. 
                    TAB CALL TAB $TYPTX<RTN>       
                    TAB DB TAB 'HELLO, WORLD',  
                    212Q<RTN> 

    HDOS 3.0 COMMON DECK CONTENTS (Cont) 
    ==================================== 
 
    TYPTX.ACM Contents (Cont) 
    ------------------------- 
 
    $TYPTX.         TAB LXI TAB H,MSG<RTN> 
                    TAB CALL TAB $TYPTX. 
    ....................................................................... 
 
 
    MOVE.ACM Contents     
    ================= 
 
    MOVE    SPACE   4,10 
    **      $MOVE - MOVE DATA 
    * 
    *       $MOVE MOVES A BLOCK OF BYTES TO A NEW MEMORY ADDRESS. 
    *       IF THE MOVE IS TO A LOWER ADDRESS, THE BYTES ARE MOVED FROM 
    *       FIRST TO LAST. 
    * 
    *       IF THE MOVE IS TO A HIGHER ADDRESS, THE BYTES ARE MOVED FROM 
    *       LAST TO FIRST. 
    * 
    *       THIS IS DONE SO THAT AN OVERLAPED MOVE WILL NOT 'RIPPLE'. 
    * 
    *       ENTRY   (BC) = COUNT 
    *               (DE) = FROM 
    *               (HL) = TO 
    *       EXIT    MOVED 
    *               (DE) = ADDRESS OF NEXT FROM BYTE 
    *               (HL) = ADDRESS OF NEXT *TO* BYTE 
    *               'C' CLEAR 
    *       USES    ALL 
 
    $MOVE   EQU     030252A                 ; IN H17 ROM 
    ....................................................................... 
 
    ECDEF.ACM Contents 
    ------------------ 
 
            SPACE   4,10 
    **      ERROR CODE DEFINITIONS. 
 
            ORG     0 
 
            DS      1           ; NO ERROR #0 
    EC.EOF  DS      1           ; END OF FILE 
    EC.EOM  DS      1           ; END OF MEDIA 
    EC.ILC  DS      1           ; ILLEGAL SCALL CODE 
    EC.CNA  DS      1           ; CHANNEL NOT AVAILABLE 
    EC.DNS  DS      1           ; DEVICE NOT SUITABLE 
    EC.IDN  DS      1           ; ILLEGAL DEVICE NAME 
    EC.IFN  DS      1           ; ILLEGAL FILE NAME 
    EC.NRD  DS      1           ; NO ROOM FOR DEVICE DRIVER 

    HDOS 3.0 COMMON DECK CONTENTS (Cont) 
    ==================================== 
 
    ECDEF.ACM Contents (Cont) 
    ------------------------- 
 
    EC.FNO  DS      1           ; CHANNEL NOT OPEN 
    EC.ILR  DS      1           ; ILLEGAL REQUEST 
    EC.FUC  DS      1           ; FILE USAGE CONFLICT 
    EC.FNF  DS      1           ; FILE NAME NOT FOUND 
    EC.UND  DS      1           ; UNKNOWN DEVICE 
    EC.ICN  DS      1           ; ILLEGAL CHANNEL NUMBER 
    EC.DIF  DS      1           ; DIRECTORY FULL 
    EC.IFC  DS      1           ; ILLEGAL FILE CONTENTS 
    EC.NEM  DS      1           ; NOT ENOUGH MEMORY 
    EC.RF   DS      1           ; READ FAILURE 
    EC.WF   DS      1           ; WRITE FAILURE 
    EC.WPV  DS      1           ; WRITE PROTECTION VIOLATION 
    EC.WP   DS      1           ; DISK WRITE PROTECTED 
    EC.FAP  DS      1           ; FILE ALREADY PRESENT 
    EC.DDA  DS      1           ; DEVICE DRIVER ABORT 
    EC.FL   DS      1           ; FILE LOCKED 
    EC.FAO  DS      1           ; FILE ALREADY OPEN 
    EC.IS   DS      1           ; ILLEGAL SWITCH 
    EC.UUN  DS      1           ; UNKNOWN UNIT NUMBER 
    EC.FNR  DS      1           ; FILE NAME REQUIRED 
    EC.DIW  DS      1           ; DEVICE IS NOT WRITABLE (OR WRITE-LOCKED) 
    EC.UNA  DS      1           ; UNIT NOT AVAILABLE 
    EC.ILV  DS      1           ; ILLEGAL VALUE 
    EC.ILO  DS      1           ; ILLEGAL OPTION 
    EC.VPM  DS      1           ; VOLUME PRESENTLY MOUNTED ON DEVICE 
    EC.NVM  DS      1           ; NO VOLUME PRESENTLY MOUNTED 
    EC.FOD  DS      1           ; FILE OPEN ON DEVICE 
    EC.NPM  DS      1           ; NO PROVISIONS MADE FOR REMOUNTING 
                                    MORE DISKS 
    EC.DNI  DS      1           ; DISK NOT INITIALIZED 
    EC.DNR  DS      1           ; DISK IS NOT READABLE 
    EC.DSC  DS      1           ; DISK STRUCTURE IS CORRUPT 
    EC.NCV  DS      1           ; NOT CORRECT VERSION OF HDOS 
    EC.NOS  DS      1           ; NO OPERATING SYSTEM MOUNTED 
    EC.IOI  DS      1           ; ILLEGAL OVERLAY INDEX 
    EC.OTL  DS      1           ; OVERLAY TOO LARGE 
    EC.LAD  DS      1           ; File is locked against delete      /3.0a/ 
    EC.FIX  DS      1           ; Device media is fixed              /3.0a/ 
    EC.ILA  DS      1           ; Illegal Load Address               /3.0a/ 
    EC.DNL  DS      1           ; Device Not Loaded                  /3.0a/ 
    EC.DNP  DS      1           ; Device Not Locked in Memory        /3.0a/ 
    EC.DFM  DS      1           ; Device is Fixed in Memory          /3.0a/ 
    EC.IDF  DS      1           ; Illegal Date Format                /3.0a/ 
    EC.ITS  DS      1           ; Illegal Time Format                /3.0a/ 
    EC.CNR  DS      1           ; System Clock Not Resident          /3.0a/ 
    EC.SDR  DS      1           ; System Disk is Reset               /3.0a/ 
    EC.LDO  DS      1           ; Line Buffer Overflow               /3.0a/ 
    EC.CUI  DS      1           ; Can't Unlink from Interrupt Vector /3.0a/ 
    EC.PNG  DS      1           ; Permission Not Given               /3.0a/ 
    ....................................................................... 

    HDOS SYMBOL VALUES  
    ================== 
 
    This section contains a list of split-octal values for the HDOS symbols 
    discussed in this document.  These values are  presented  as  a  double 
    check,  so  you  can  compare  them  to  the  values generated when you 
    assemble the common decks.  Once again, it is important  that  you  use 
    the  common  decks and use symbolic values, rather than using the octal 
    values directly.   
     
    ECDEF Symbol Definitions  
    ------------------------ 
     EC.CNA = 000004A           EC.FNO = 000011A          EC.NCV = 000050A 
    *EC.CNR = 000064A           EC.FNR = 000034A          EC.NEW = 000021A 
    *EC.CUI = 000067A           EC.FOD = 000043A          EC.NOS = 000051A 
     EC.DDA = 000027A           EC.FUC = 000013A          EC.NPM = 000044A 
    *EC.DFM = 000061A           EC.HIN = 000000A          EC.NRD = 000010A 
     EC.DIF = 000017A           EC.1CN = 000016A          EC.NVM = 000042A 
     EC.DIW = 000035A          *EC.IDF = 000062A          EC.OTL = 000053A 
     EC.DNI = 000045A           EC.IDN = 000006A          EC.RF  = 000022A 
    *EC.DNL = 000057A           EC.IFC = 000020A         *EC.SDR = 000065A   
    *EC.DNP = 000060A           EC.IFN = 000007A          EC.UNA = 000036A 
     EC.DNR = 000046A          *EC.ILA = 000056A          EC.UND = 000015A 
     EC.DNS = 000005A           EC.ILC = 000003A          EC.UUN = 000033A 
     EC.DSC = 000047A           EC.ILO = 000040A          EC.VPM = 000041A 
     EC.EOF = 000001A           EC.ILR = 000012A          EC.WF  = 000023A 
     EC.EOM = 000002A           EC.ILV - 000037A          EC.WP  = 000025A 
     EC.FAO = 000031A           EC.IOI = 000052A          EC.WFV = 000024A 
     EC.FAP = 000026A           EC.IS  = 000032A                           
    *EC.FIX = 000055A          *EC.ITS = 000063A                           
     EC.FL  = 000030A          *EC.LAD = 000054A                           
     EC.FNF = 000014A          *EC.LBO = 000066A                           
      
    * New for HDOS 3.0 
    ..................................................................... 
 
    HOSDEF Symbol Definitions  
 
    -------------------------                                              
                                                                            
     .CHFLG  =  000060A        .EXIT  =  000000A        .PRINT  =  000003A 
     .CLEAR  =  000055A       *.GDA   =  000013A        .READ   =  000004A 
     .CLEARA =  000045A        .LINK  =  000040A        .RENAME =  000051A 
     .CLOSE  =  000046A       *.LOADD =  000062A        .RESET  =  000204A 
     .CLRCO  =  000007A        .LOADO =  000010A       *.RESMNS =  000250A  
     .CONSL  =  000006A       *.LOG   =  000177A        .SCIN   =  000001A 
    *.CRC16  =  000013A        .MONMS =  000202A        .SCOUT  =  000002A 
     .CTLC   =  000041A        .MOUNT =  000200A        .RESMNS =  000250A 
     .DAD    =  000207A        .NAME  =  000054A        .SETTOP =  000052A 
     .DECODE =  000053A        .OPENC =  000045A       *.TASK   =  000101A 
     .DELETE =  000050A        .OPENR =  000042A       *.TDU    =  000122A 
     .DMNMS  =  000203A        .OPENU =  000044A        .VERS   =  000011A  
     .DMOUN  =  000201A        .OPENW =  000043A        .WRITE  =  000005A 
     .ERROR  =  000057A        .POSIT =  000047A 
                                                                            
      CO.FLG  =  000001A        CS.FLG  = 000200A        CSL.CHR =  000001A 

    HDOS SYMBOL VALUES  
    ================== 
 
    HOSDEF Symbol Definitions (Cont) 
    -------------------------------- 
 
      CSL.ECH =  000200A        CSL.WRP = 000002A        CTP.2SB =  000010A 
      CTP.BKM =  000002A        CTP.BKS = 000200A        CTP.ML1 =  000040A 
      EC.DDA  =  000027A        EC.DIF  = 000017A        EC.DIW  =  000035A 
        
      EC.CNA  =  000004A        EC.FOD  = 000043A        EC.NRD  =  000010A 
      EC.CNR  =  000064A        EC.FUC  = 000013A        EC.NTF  =  000311A 
      EC.CUI  =  000067A        EC.ICN  = 000016A        EC.NTM  =  000003A 
      EC.DDA  =  000027A        EC.IDF  = 000062A        EC.NVM  =  000042A 
      EC.DFM  =  000061A        EC.IDN  = 000017A        EC.OTL  =  000053A 
      EC.DIF  =  000017A        EC.IFC  = 000020A        EC.PNG  =  000070A 
      EC.DIW  =  000035A        EC.IFN  = 000007A        EC.RF   =  000022A 
      BC.DNI  =  000045A        EC.ILA  = 000056A        EC.SDR  =  000065A 
      EC.DNL  =  000057A        EC.ILC  = 000003A        EC.TAA  =  000301A 
      EC.DNP  =  000060A        EC.ILO  = 000040A        EC.TCA  =  000305A 
      EC.DNR  =  000046A        EC.ILR  = 000012A        EC.TIF  =  000304A 
      EC.DNS  =  000005A        EC.ILV  = 000037A        EC.TNA  =  000302A 
      EC.DSC  =  000047A        EC.IOI  = 000052A        EC.TNF  =  000307A 
      EC.EOF  =  000001A        EC.IS   = 000032A        EC.TSN  =  000036A 
      EC.EOM  =  000002A        EC.ITF  = 000300A        EC.TUN  =  000303A 
      EC.FAO  =  000031A        EC.ITS  = 000063A        EC.UNA  =  000036A 
      EC.FAP  =  000026A        EC.LAD  = 000054A        EC.UND  =  000015A 
      EC.FIX  =  000055A        EC.LB0  = 000066A        EC.UUN  =  000033A 
      EC.FL   =  000030A        EC.NCV  = 000050A        EC.VPM  =  000041A 
      EC.FNF  =  000014A        EC.NEM  = 000021A        EC.WF   =  000023A 
      EC.FNO  =  000011A        EC.NOS  = 000051A        EC.WP   =  000025A 
      EC.FNR  =  000034A        EC.NPM  = 000044A        EC.WPV  =  000024A 
 
                                I.CONFL = 000004A        I.CONTY =  000001A 
      I.CONWI =  000003A        I.CSLMD = 000000A        I.CUSOR =  000002A 
      S.DATC  =  040310A        S.DATE  = 040277A        S.HIMEM =  040316A 
      S.OMAX  =  040324A        S.SYSM  = 040320A        S.USRM  =  040322A 
 
    * New for HDOS 3.02 
    ....................................................................... 
 
    HOSEQU Symbol Definitions  
    ------------------------- 
    USERFWA = 42200A 
    STACK   = 42200A 
 
    CO.FLG  = 000001A          CTP.MLO = 000020A         I.CUSOR = 000002A 
     
    CS.FLG  = 000200A          CTP.TAB = 000001A         S.DATC  = 040310A 
    CSL.CHR = 000001A          CTP.2SB = 000010A         S.DATE  = 040277A 
    CSL.ECH = 000200A          I.CONFL = 000004A         S.HIMEM = 040316A 
    CSL.WRP = 000002A          I.CONTY = 000001A         S.OMAX  = 040324A 
    CTP.BKM = 000002A          I.CONWI = 000003A         S.SYSM  = 040320A 
    CTP.BKS = 000200A          I.CSLMD = 000000A         S.USRM  = 040322A 
    CTP.MLI = 000040A 
    ************************************************************************ 

                             PART 7 - PROLOGUE.SYS 
                             +++++++++++++++++++++ 
 
    A PROLOGUE.SYS is a special program generated in Assembly Language. 
 
    Appendix  13-A  contains  an example of the HDOS "PROLOGUE.SYS" feature 
    incorporated into assembly language source code. 
 
    When  HDOS  is started at the BOOT command, it will attempt to transfer 
    control to SY0:PROLOGUE.SYS.  If such a file cannot be  found,  control 
    is  transferred  to  "SYSCMD.SYS."   The  use of a PROLOGUE.SYS sets up 
    special automatic tasking parameters at bootup time,  such  as  loading 
    device  drivers,  or  starting  menu  programs  in  MBASIC  or Assembly 
    Language.  If the  processing  of  your  PROLOGUE.SYS  program  is  not 
    desired  at  boot  time,  just  simply  type  IGNORE  when HDOS states: 
    <BOOT><B>?  instead of the usual <RTN>. 
    *********************************************************************** 
 
                         PART 8 - PROGRAMMING EXAMPLE 
                         ++++++++++++++++++++++++++++ 
 
    Refer to pages 13-72 thru 13-76 for typical programming examples. 
 
                                 ** CAUTION ** 
              The  following  files,  being  typical  source code 
              files, extend beyond the  normal  80  column  line! 
              They  cannot  be either viewed by the standard text 
              editor, nor can they be printed at the normal 10 or 
              12  CPI.   Therefore,  you  must  take the measures 
              indicated below. 
 
    There are two options: (1) This material may be monitored on the screen 
    using an HDOS editor such as EDIT19 which permits viewing of  the  file 
    contents  extending  beyond  80  columns.   (2)  You  can print out the 
    following files on a line printer, but you must use a 17 CPI instead of 
    the customary 10 or 12 CPI. 
               
    (1) Screen Option: 
    ------------------ 
    To  scroll  the  screen so that it goes off the left side, while in the 
    SCREEN mode of EDIT19, Version 3.10, hit the ENTER key  and  type  LEFT 
    60<RTN>.   This  will cause your screen to move 60 columns to the left, 
    and you will be able to see the hidden part of the listing.  To  return 
    the screen to normal, once more hit the ENTER key and type LEFT 0<RTN>. 
    If your version of EDIT19 is earlier than 3.10, substitute the  command 
    "DISCARD" for "LEFT" as indicated above.  The same technique is used. 
 
    (2) Print Option: 
    ----------------- 
    Set  your  printer  according  to  the  following settings and print as 
    usual.  You will receive a smaller printout,  but  the  entire  listing 
    will be visible on the page. CPI=17, LPI=8, FORM=88, PAGE=80, LMARG 15. 
    *********************************************************************** 
 
 
 




